

ISPF Developer Tips and Tricks

```
Menu Utilities Compilers Options Status Help
ISPF Primary Option Menu
Option ==> █
0 Settings      Terminal and user parameters
1 View          Display source data or listings
2 Edit          Create or change source data
3 Utilities     Perform utility functions
4 Foreground   Interactive language processing
5 Batch         Submit job for language processing
6 Command       Enter TSO or Workstation commands
7 Dialog Test  Perform dialog testing
9 IBM Products  IBM program development products
10 SCLM         SW Configuration Library Manager
11 Workplace   ISPF Object/Action Workplace
12 z/OS System z/OS system programmer applications
13 z/OS User    z/OS user applications

Enter X to Terminate using log/list defaults

<  Calendar  >
      March   2019
Su Mo Tu We Th Fr Sa
          1  2
3  4  5  6  7  8  9
10 11 12 13 14 15 16
17 18 19 20 21 22 23
24 25 26 27 28 29 30
31
Time . . . . : 11:21
Day of year. :  079
```

Contents

Dedication.....	5
Caveats and Disclaimer	5
Where to Find Updates	5
How to Contribute	6
Revision History	7
Contributors	8
Authors/Editors/Contributors.....	8
Contributors who may not know it.....	8
Introduction	8
Sample PDS.....	9
General Comments	10
Installing Your Application	11
Making the Application Available	11
Sample Stub	11
Sample PLP Definition	12
Adding to an Existing ISPF Menu	12
Add to the ISPF Commands Table	12
Dynamic ISPF Menus.....	13
Disabling Keylists	13
ISPF Panels.....	14
Testing Panels.....	14
Tutorial Panels	14
Scrolling Panels.....	15
Field Level Help.....	16
Using Point and Shoot (PNS) with ISPF Panels	17
PopUp Panels	20
Dynamically Turning Off PFSHOW	21
Progress Popup Panels.....	22
Action Bars and Pull-downs.....	24
Panel REXX.....	25
Basic Example	25
Verify a Data Set Name within Panel REXX.....	26
Panel Scrolling Fields	27
Dynamic Areas	29
ISPF Skeletons	33

Simple Skeleton	34
Skeleton with REXX	35
Using the TSO Stack	35
Using TSO Commands	36
ISPF Tables	37
Find.....	37
Enabling RFIND.....	39
Locate a Row.....	41
Selecting Multiple Rows for Processing	42
Full Example	43
ISPF Messages.....	54
ISPF Edit Macros	55
Symbolic Handling.....	55
Centering Text using an Edit Macro.....	57
Invoking ISPF Edit with a Macro	58
Passing Data to an Edit Macro	59
Invoking an Edit Macro on All Members of a PDS	60
Changing ISPF Edit Commands with Macros	61
Library Services	63
LMDLIST – List Data Sets	63
Miscellaneous	65
Browsing Data in a REXX Stem.....	65
ISPF in Batch	66
Randomize DDname and Table Names	67
Based on the REXX Exec Name.....	67
Using the REXX Random Function	67
Another Random String.....	67
Getting the DDname from the System.....	68
Stem Sort	68
Full Stem Sort example.....	69
More on Stem Sorting using BPXWUNIX.....	69
Converting the User Provided Data Set Name to a Full Data Set Name	70
Default ISPF Terminal Type	70
Useful Tools	70
ISRDDN.....	70
ISPLIBD.....	70

ALTLIB Display	70
Debugging Hints/Tips.....	70
Displaying the ISPF Panel Name	70
ISPF Dialog Test.....	70
REXX Trace.....	71
Miscellaneous Tips	71
Appendix.....	73
Useful Tools	73
CMT.....	73
LOADISPF/DROPISPF	73
REXXFORM	74
STEMEDIT.....	74
TRYIT	75
Other Tools of Note (meaning they are worth checking out)	77
PDS (the Swiss Army Knife of Utilties)	77
PDSEGEN	77
ISPF CMD.....	77
Useful Websites	79

Dedication

This document is dedicated, with extreme thanks, to those who came before us. To the Blue Berets who contributed and managed the MVT Mods Tapes, to those who freely shared their code and documentation via cards, tapes, Gopher, listservs, and then the Web, to those who contribute to the CBTTape, to those who speak at SHARE and other organizations sharing their experience and knowledge. And to all the IBMers who share via the various listservs, forums, and PMRs. We are all better developers because of the help of others, whether it is overt from a document or presentation, or by allowing us to learn from their code. This document is released in the spirit that you, our readers, will continue this tradition.

Caveats and Disclaimer

This document and the code samples are provided without warranty or guarantee. If you use any of the code for production purposes, you should thoroughly test and validate it before use.

There is no copyright associated with the sample code so feel free to use any/all of it within your own code (after testing and validation of course). The term, YMMV (your mileage may vary) applies.

Where to Find Updates

Updated versions will be posted to www.lbdsoftware.com, and www.cbttape.org in file 990.

How to Contribute

This document is intended to be a living document and that requires regular updates. If you would like to provide sample code, and the prose to go with it, please e-mail ispfdev.tips@gmail.com and one of the volunteer team members will acknowledge your communication.

Contribution guidelines:

1. Code should follow the sample naming conventions:
 - a. PNxxx for Panels
 - b. RXxxxx for REXX
 - c. SKxxx for Skeletons
2. Provide it in TSO Transmit format to prevent EBCDIC/ASCII conversion issues during transport.
 - a. Put the elements in a PDS
 - b. Issue the TSO command: XMIT x.y DS(your.pds) OUTDS(contribution.XMIT)
 - i. Note x.y can be anything
 - c. Binary download and e-mail the contribution.XMIT data set
3. Prose may be plain text or n Word format.
4. Suggest where in the document to place it.
5. Examples should be generic and not depend upon products or tools that are not available on all systems (that is – don't assume every site has product xyz).

There is no guarantee that any contributions will be used – that is left to the discretion of the editors and will mostly depend on their available time.

Updates to this document and the related samples data set will be announced on the ibm-main listserv (join/view at <https://listserv.ua.edu/archives/ibm-main.html>) and posted at www.lbdsoftware.com.

Revision History

Version 1.1	31 May 2019	<ol style="list-style-type: none">1. Several minor corrections from W. Jensen along with a new randomized example2. Change date in footer3. Sample Skeleton REXX with TSO services from W. Jensen4. Sample using BPXWDYN for a random ddname from W. Jensen.5. Section on using LISTDSI to return a fully qualified data set name.6. Additional sample random routines for ddname or table names.7. Add info on PANELID ON/OFF8. Update the How to Contribute section9. Added a chapter on Other Tools10. Added section on Stem Sorting from John Kalinich.
Version 1.0	18 April 2019	Initial release

Contributors

Authors/Editors/Contributors

Lionel B. Dyck	John Kalinich	Bruce Koss
Willy Jensen		

Contributors who may not know it

Doug Nadel via his Presentation ISPF Panels Beyond the Basics which can be found at https://sites.google.com/site/schlabb/home/hints-tips/ispf	Thomas Conley	John McKown via IBM-Main
Bill Godfrey (via IBM-Main)	Albert Ferguson via IBM-Main	

Introduction

This document is intended to provide the ISPF application developer, whether full time or ad hoc, with tips and tricks to exploit the capabilities of ISPF within the z/OS environment. There are a few tools that will be mentioned in the tips that the contributors feel will be useful. See the [Appendix](#) for details on those tools.

Become familiar with the ISPF Edit MODEL command. This great feature will provide sample code for ISPF Panels, Skeletons, REXX, and more.

Download, or bookmark, the ISPF publications from IBM for the version of z/OS that you will be working with. Given the level of ISPF development, it is unlikely that there will be new features or functions added to newer versions –one never knows, however one can hope.

It is expected the reader understands REXX and ISPF to take full advantage of this document.

To accompany this document, a PDS has been created with samples for many of the topics discussed. This PDS is distributed in TSO Transmit (XMIT) format which must be processed by the TSO RECEIVE command after being uploaded to z/OS using a binary transfer method into a sequential data set with RECFM=FB and LRECL=80.

```
RECEIVE INDS(upload.xmit)
```

The topics with examples below will have a box with the example member names that looks like this:

Member-name

Note that the sample code may include additional comments from what is included in the examples in this document.

Sample PDS

Within the Sample PDS are members that contain sample ISPF Panels, Skeletons, and sample REXX code. All are fully functional for demonstration purposes.

Member \$\$README provides additional information about the samples and member \$\$XMIT provides information on the two members in TSO Transmit format.

Included is an ISPF dialog that demonstrates many of the ISPF table handling function using the Sample PDS.

To use the dialog:

1. go into ISPF 3.4 and list the Sample PDS data set name
2. Then enter M to display the Member List
3. Scroll down to \$DEVISPF
4. and enter EX next to it to begin the dialog which displays an ISPF table of the members and supports several commands and row selection options:

```
----- ISPF Developer Tips/Tricks Examples -- Row 1 to 16 of 47
Command ==> █                               Scroll ==> CSR

Commands:  Find xx Locate xx Only xx  R Refresh
Selections: B Browse E Edit R Receive X eXecute

Sel Member  Type  Description
-  $README  Text  What you are reading
-  $XMIT     Text  Doc on how to process the XMIT members
-  $DEVCPY   REXX  Exec using LMCOPY to copy Samples into User Libraries
-  $DEVCPYP  Panel ISPF Prompt Panel for $DEVCPY
-  $DEVISPF  REXX  Exec to display the Sample Members
-  $DEVPH    Panel ISPF Tutorial Panel for $DEVPP
-  $DEVPP    Panel ISPF Panel used by $DEVISPF
-  $DEVPIX   Panel ISPF Popup Panel to Prompt for parms when executing REXX
-  #RXFORM   XMIT  #RXFORM tool in TSO Transmit format
-  #TRYIT    XMIT  #TRYIT tool in TSO Transmit format
-  CMT       Macro REXX Edit Macro to insert Comments in Code/JCL/etc.
-  JCBAT1    JCL   Sample ISPF in Batch JCL using the TASID Utility
-  JCBAT2    JCL   Sample ISPF in Batch JCL using the RXLMD REXX
-  LOADISPF  REXX  REXX code to be used with other code
-  PNABC     Panel Sample ISPF Panel to demonstrate an Action Bar
-  PNFLDH    Panel ISPF Panel to demonstrate Field Level Help
```

Hint: Execute (x) the \$DEVCPY exec to copy these samples from the sample library into your own SYSEXEC and ISPLIB libraries so that you can make changes, experiment, and learn.

General Comments

When writing code, it is important that others be able to read, understand, and if necessary, update (fix) the code. The best way to do that is to follow coding practices that have been developed over the past several decades:

1. Use meaningful variable names where possible. ISPF variables are limited to 8 characters, while REXX variables can be longer.
 - a. ISPF variable: &username
 - b. REXX variable: TSO_User_Name
2. Document the code with comments
 - a. There are many styles of comments and the best one is what the developer (you) are most comfortable in using. Some like to put comments on the lines of code, while others like to put comment blocks before code sections.
 - b. An ISPF Edit macro, [CMT](#), mentioned in the Appendix can make it significantly easier to enter comments. It creates block comments and supports entering comments from the command line or via a popup panel.
 - c. REXX code supports comments – so do ISPF Messages, Panels, and Skeletons. That means that you can add useful comments in all those elements so that others will know what you were thinking when you developed them.
 - d. Not everything requires comments. Too many comments can get in the way of the code. Be judicious and not verbose.
3. If there is a complex way to write code and a simple way, choose the simple way. You never know who may have to pick up your code in 5, 10, 20 years and fix something that you never anticipated.
4. Where possible, use standard system routines and interfaces.

Installing Your Application

Once you develop your ISPF application the next decision is how to make it available to your users. The first thing to consider is how will the application be executed, or how will the elements be accessed. The second thing to consider is how will the user find the application so that they can use it.

Making the Application Available

Once your application is completed, the various elements (execs, panels, programs, skeletons, messages, tables, etc.) need to be made available to the users.

There are three ways to do this:

1. Copy the elements into libraries that are already allocated to the users ISPF Logon (via Logon Proc or Logon Clist).
2. Write a stub Exec that performs the dynamic allocations and then invokes the application which must be installed in a library in the users SYSEXEC or SYSPROC allocations.
3. Use the [Product Launch Point](#) (PLP) discussed below.

Sample Stub

This is an example to invoke the DFSORT ISPF dialog and it is from Tom Conley's Dynamic ISPF Starter Set package which can be found in File 495 at www.cbttape.org where you can find samples for you to use for over 100 different software packages.

```
/* rexx */
/* trace i */
/*****
/* This exec invokes IBM's DFSORT dialog. It assumes that
/* SYS1.SICELINK is available via LINKLIST. If not, add this
/* loadlib to the LINKLIST, or change this exec to use a Dynamic
/* STEPLIB product.
/*****
parse arg zcmd
address tso "ALTLIB ACT APPL(CLIST) DA('SYS1.SICECLIB') "
address ispexec "LIBDEF ISPMLIB DATASET ID('SYS1.SICEMENU') STACK"
address ispexec "LIBDEF ISPPLIB DATASET ID('SYS1.SICEPENU') STACK"
address ispexec "LIBDEF ISPLIB DATASET ID('SYS1.SICESLIB') STACK"
address ispexec "LIBDEF ISPTLIB DATASET ID('SYS1.SICETLIB') STACK"
address ispexec "SELECT PGM(ICEPMD01) NOCHECK PARM("zcmd")",
                "SCRNAME(DFSORT) "
address ispexec "LIBDEF ISPMLIB"
address ispexec "LIBDEF ISPPLIB"
address ispexec "LIBDEF ISPLIB"
address ispexec "LIBDEF ISPTLIB"
address tso "ALTLIB DEACT APPL(CLIST) "
```

Sample PLP Definition

This is how that would look in a PLP definition:

```
----- Product Launch Point Management -----
Command ==>

Enter, Verify, or Change (* = required)

*Application Name      SYNCM___ A unique application name 1-8 characters
Application ID        SX21     1-4 characters for application id
*Description          Syncsort ISPF Messages Dialog_____
Step Library
ISPF Load Library     'SYS2.SYNC.MFX210.SYNCLoad'_____
ISPF Message Dataset  'SYS2.SYNC.MFX210.SYNCLIB'_____
ISPF Panel Dataset    'SYS2.SYNC.MFX210.SYNCLIB'_____
ISPF Skeleton Dataset _____
ISPF Table Dataset    _____
Clist Library         _____
REXX EXEC Library     _____
*Application Start (select one)
  Command
  Program      SS21MSG3 Parm _____
  ISPF Panel   _____ Panel option _____
Additional Datasets: No_ Yes to add more
```

Adding to an Existing ISPF Menu

The application can be added to an existing ISPF Selection Panel (menu) where it can be found by those users who are aware of that Panel. In some shops, that means adding the application to `ISR@PRIM` as one more in a long list of ISPF applications. Others will add the application of a menu that is accessible from `ISR@PRIM`.

The advantage is that your application is now on a Selection Panel and thus available to your intended target users.

The disadvantage is that your application is on a Selection Panel that your intended target users, including new users, may not be aware of. If they are aware of the Panel, they may not be aware that they must scroll down to find your application.

Another challenge is making sure that all copies of the Selection Panel are updated, this is especially true for `ISR@PRIM` and `ISP@MSTR`, for which there may be multiple versions in the various libraries in the `ISPPLIB` allocation, and all dependent upon the order they are allocated.

Add to the ISPF Commands Table

This provides a fast path to your application from any ISPF Command line. This is an excellent option providing the application name is unique (had one ISPF application that was installed that had a command that was the same as a command in the ISPF Command Table and that caused a lot of confusion until it was researched).

There is also the challenge for existing, and new, users who need to find the name of the command to invoke the application – like adding the application to a scrolling, or sub, menu Selection Panel.

Dynamic ISPF Menus

This is a plug for a tool called the Product Launch Point (PLP), which is an open source dynamic ISPF menu. This tool can be found at <http://lbdsoftware.com/ispftools.html>, and at <http://www.cbttape.org> in file 312.

PLP is table driven ISPF menu. Add the command PLP to the ISPF commands table and your user has immediate access to a searchable, scrollable, flexible ISPF menu.

PLP supports multiple menus based on how it is invoked, and it supports sub-menus.

The advantages are numerous, among them are:

1. Easily add, or update, an application.
2. When adding an application PLP allows defining libraries that will be
 - a. STEPLIB'd (if a dynamic STEPLIB command is available)
 - b. LIBDEF'd for ISPF libraries
 - c. ALTLIB for CLIST and REXX libraries
 - d. Specify an Application ID
 - e. Invoke a Panel, Program, REXX/Clist – and pass parms

PLP is used at a number of large, and small, z/OS installations very successfully as they have found it greatly reduces the efforts to manage and maintain menus for access to applications, while at the same time making it very easy to find and execute the installed applications.

Disabling Keylists

Some sites have disabled application keylists via the ISPF Configuration Table, while others allow them. If you want to disable them for the period of your dialog you can use this code:

```
"VGET (ZKLUSE) PROFILE" /* Obtain Keylist */
SAVE_ZKLUSE = ZKLUSE /* Save Keylist */
ZKLUSE = "Y" /* Keylist on */
"VPUT (ZKLUSE) PROFILE" /* Set Keylist */
... code...
ZKLUSE = SAVE_ZKLUSE /* Restore */
"VPUT (ZKLUSE) PROFILE" /* Set Keylist */
```

ISPF Panels

Testing Panels

When developing ISPF Panels, it is easy to make a typographic error in the panel code. Using ISPF Test can be a challenge (but one you should take the time to learn). However, there is a tool included with the samples call [TRYIT](#) that can be used to validate your ISPF Panels easily and while you're still in ISPF Edit (you don't have to save the member to test it).

The TRYIT command syntax that you will use for ISPF Panels is:

TRYIT	For any Panel
TRYIT TUT	For Tutorial Panels
TRYIT POP	For Popup Panels

If there are any errors then there will be a short message indicating there is an issue and pressing F1 will display the longer, more detailed message with more information.

See the Appendix for more information on [TRYIT](#).

Tutorial Panels

Most, not all, ISPF panels should have a tutorial behind it. The tutorial is accessed using the HELP command (typically assigned to the F1 key).

Some things to always consider with Tutorial panels:

1. If you have multiple Tutorial panels, then connect them using the &ZUP and &ZCONT commands.
2. If you only have one Tutorial panel then set &ZCONT to the panel so that when the user presses the enter key, they will remain on that panel.
3. With multiple Tutorial panels:
 - a. The 1st always has &ZCONT referencing the next panel
 - b. The 2nd thru the last has &ZUP referencing the previous panel and &ZCONT referencing the next
 - c. The last should have &ZCONT referencing the 1st Tutorial panel.
4. Try to keep the Tutorial panel body to 23 (or less) records. This is to allow the Tutorial to be viewed on a 3270 model 2 terminal (this configuration supports 24 rows, but you should allow for at least 1 row for an ISPF Split Screen line.
5. All the Tutorial panels should have the same general layout for consistency.
6. The user can 'scroll' between Tutorial panels using function keys F10 (Previous) and F11 (Next).
7. For complex dialogs consider a Tutorial Menu panel.
 - a. Use the ISPF Edit MODEL command
 - b. Select F0 (PANFORM)
 - c. Select F5 (TUTORIAL)

Scrolling Panels

PNAREA and RXAREA

There are times that the data on a panel is more than the depth of the user screen and rather than creating multiple panels (the recommended technique), it is easy to define a panel area that allows scrolling.

Note that this will work for data entry panels, tutorial panels, and popup panels (table panels are already scrollable).

This sample REXX will display the following panel until the F3 (END) is pressed.

```
/* rexx */
Address ISPEXec
do forever
  'display panel(pnarea) '
  if rc > 0 then leave
end
```

This is part of the Panel:

```
)attr default(%+_)
} area(SCRL) Extend(ON)
)Body Expand(\\)
%Sample -\\-(~Scrolling Panel Demo)\\- Sample
%Command ==>_ZCMD
+
+ Sample of a Scrolling Area - use &zpf07 and &zpf08 keys to scroll +
+ And then%F3+to Exit.
%-----}
}help -----}
)Area Help
+With thanks to Mr. John Kalinch for allowing us to 'borrow' the FAQ for the
+PDS 8.6 command (found on the CBTape File 182 at www.cbttape.org)
+
+ PDS Command Processor FAQ
. . .
)Init
)PROC
)END
```

Note that there is a size limit, which I suspect is byte related, but for this example at 1000 records in the area it works and with a few more it fails (at 1023). If you have that many records it may be better to use ISPF Browse (or Edit or View) and place the data into a data set.

Be aware that when pressing ENTER the panel will automatically scroll back to the top.

Field Level Help

PNFLDH, PNFLDH1, PNFLDH2, and RXFLD

For some applications having more detailed information for specific fields on the panel is helpful. IBM has provided the ability to have Field Level Help panels that make this very easy by allowing the user to place the cursor on a specific entry field and then press F1 to display a specific Help panel.

Here is a sample data entry panel with Field Level Help:

```
)ATTR DEFAULT(%+_ )
$ TYPE(INPUT) INTENS(LOW) hilite(uscore)
)BODY Expand(\\)
+\\-%Sample Panel with Field Level Help+\\-\\
%Command ==>_zcmd
%
+   Place the cursor on any of the fields below and press F1 to
+   view the Field Level Help.
+
+   Field one: $z      +           Sample Input Field
+   Field two: $z      +           Sample Input Field
+
+   +Press%F3+when ready to exit.
)INIT
.zvars = '(field1 field2)'
)PROC
)Help
Field(field1)  panel(pnflhd1)
Field(field2)  panel(pnflhd2)
)END
```

The Field Level Help is defined in the)HELP section of the panel using the FIELD statements, where the field name, enclosed in parenthesis, is the entry field on the panel.

This is one of the sample Field Level Help panels. Notice this is a normal popup panel, although field level help does not require a popup, that is the typical usage.

```
)ATTR DEFAULT(%+_ )
)BODY WINDOW(45,4)
+
+ This is a sample help for%Field One
+
+(Press%F3+to close)
)INIT
&ZWINTTL = 'Field Level Help One'
)END
```

And this rexx code can be used to display and experiment with this capability:

```
Address ISPExec
do forever
  'Display Panel(pnflhd)'
  if rc > 0 then leave
end
```

Using Point and Shoot (PNS) with ISPF Panels

PNPNS, RXPNS, and RXPNSL

How to setup point and shoot in ISPF panel table headers

There are times where an ISPF panel needs to support point and shoot logic and this is a simple way to do it.

There are two approaches to doing this. But if the user does not have tab to Point-and-Shoot enabled then some of the usefulness is lost. Tab to Point-and-Shoot can be enabled using the ISPF command ISPFVAR PSTAB(ON).

One approach is using TYPE(PS) which allows no other attributes.

The other is TYPE(OUTPUT) with PAS(ON) which does allow most other attributes to be used such as intens, color, caps, justification, and hilite for the field.

Define some attributes

```
# type(output) intens(high) caps(off) just(left) pas(on) hilite(uscore)
```

```
] type(output) intens(high) caps(off) just(asis) pas(on) hilite(uscore)
@ type(PS)
```

Use the appropriate attribute based on the justification. In some cases, you will want to use left justification and in others, you need specific placement so asis justification is what you need. These fields are enabled for overtyping but there is no effect from doing so.

If using TYPE(PS) then you are not able to define any other attributes, but the advantage is that these fields are not enabled for overtype as the TYPE(OUTPUT) fields are.

Use them in the table header

```
#Name+ Gen Abs#CRdate ]MDdate +V.M #Size+
```

In this example both left and asis justification are used. Some of the headers are not defined for point and shoot (e.g. GEN Abs, V.M). The rest are defined for left justification except the MDdate which is defined using asis justification.

Or use them within a panel

Enter value: _val#xyz+

Define the output variable values in the)INIT section

```
&name = 'Name'
&crdate = 'Created'
&size = 'Size'
&xyz = 'XYZ'
```

As you can see all of the variable names match the output name in the header row. The &mddate has leading and trailing blanks so that it fills out the space in the header, so the fields are aligned where they need to be.

1. Define the Point and shoot

```
) PNTS
FIELD(name)      VAR(ZCMD) VAL('SORT NAME')
FIELD(crdate)    VAR(ZCMD) VAL('SORT CREATED')
FIELD(mddate)    VAR(ZCMD) VAL('SORT CHANGED')
FIELD(size)      VAR(ZCMD) VAL('SORT SIZE')
FIELD(xyz )      VAR(ZCMD) VAL('SET XYZ')
FIELD(ZPS00001)  VAR(ZCMD) VAL('Something')
```

In the) PNTS (point and shoot) section the syntax is:

FIELD(variable name)

This defines the variable of a point and shoot field in the ISPF display.

If using TYPE(PS) then the field name must be ZPSnnnnnn where nnnnnn is 000001 to the number of TYPE(PS) fields on the panel and go from left to right and top to bottom.

VAR(variable)

This defines the variable to be set when the point and shoot field is selected. In this case ZCMD will be updated and passed to the application.

VAL(value)

The value is the text that will be inserted into the variable defined in VAR(). Enclose in quotes if more than one word or if spaces are included in the text.

And assuming your dialog understands the zcmd values you're good to go.

One way to do this so that the panel gets updated but the processing code doesn't process is this technique which assumes that the point and shoot field sets ZCMD to a value of 'SET XYZ' and that causes the code to set panel variable to 'X', reset zcmd to null, and then iterate which redisplay the panel.

Here is an example ISPF Panel to demonstrate the two types of Point and Shoot fields and how they work.

Use this REXX Exec to test the following ISPF Panel:

```
/* REXX */
Address ISPEXec
do forever
  'Display Panel(pnpns) '
  if rc = 8 then call done
  if rc > 8 then do
    say zerrsm
    say zerrlm
  end
  exit
end
if translate(zcmd) = 'EXIT' then leave
end
exit
```

And this is the ISPF Panel with a name of PNPNS

```
)ATTR DEFAULT(%+_ )
$ type(ps)
# type(output) caps(off) just(left) color(red) hilite(uscore)
@ type(output) caps(off) pas(on) hilite(uscore) color(yellow)
} type(output) caps(off) pas(on) hilite(uscore) color(blue)
)body
%Command ==>_zcmd          +
+
%Test:  $One+      $Two+
+      $Three+    $Four+
+      @Five+     @Six+
+
%Value #result            +
+
+Click here when ready to)Exit+
+
+Tab to any of the point and shoot fields and press%Enter+or move the
+mouse point to them and double click. See the results in the%Value+field.
+Note that the%Five+and%Six+fields are point and shoot fields but you can
+change the text that displays in them with a drawback that you can
+actually type into those fields with zero affect.
+
)Init
&five = 'Five'
&six  = 'Six'
&exit = 'Exit'
)Proc
)PNTS
FIELD(ZPS00001) VAR(RESULT) VAL('Value of 1')
FIELD(ZPS00002) VAR(RESULT) VAL('We have two')
FIELD(ZPS00003) VAR(RESULT) VAL('Now three')
FIELD(ZPS00004) VAR(RESULT) VAL('Four')
FIELD(Five)     VAR(RESULT) VAL('Five or 5')
FIELD(Six)      VAR(RESULT) VAL('Six or 6')
FIELD(Exit)     VAR(zcmd)   VAL(Exit)
)end
```

PopUp Panels

PNPOP and RXPOP

When writing ISPF dialogs there are times when you want to prompt the user, or just provide some feedback, but you don't want to take a full window to do it. That is where a popup can be very helpful.

A Popup Panel is defined on the `) BODY` statement using the `WINDOWS (ww, hh)` field where `ww` is the number of columns wide, and `hh` is the number of rows deep.

Here is a sample panel (PNPOP):

```
)ATTR DEFAULT(%+_)
 @ type(output) caps(off) just(left)
)BODY WINDOW(45,8)
+
%This is a sample POPUP panel
  +to demonstrate the capabilities of it.
+
+and how to add a variable of the userid
+
  %userid:@zuser
+
)INIT
&ZWINTTL = 'Sample Popup Panel'
)END
```

To keep it simple the variable `&zuser`, an ISPF variable, is used to demonstrate a variable.

The size of the popup panel is 45 characters wide and 8 rows deep.

To 'spiff' up the panel we have added a panel title using the `&ZWINTTL` variable in the `) INIT` section to define a title to be used on the popup panel.

To display the popup panel, use the following code:

```
/* REXX */
Address ISPEXEC
'Addpop Row(4) Column(6) '
'Display Panel(pnpop) '
Save_Rc = rc
'Rempop'
```

The `ADDPOP` command supports optional parameters. The `ROW` keyword defines which row on the screen to display the popup, and the `COLUMN` how many characters from the left to begin the display. In this case we have the panel being displayed starting on row 4 and column 6. If `ROW` and `COLUMN` are not provided, then the popup will be at the top left of the display.

A popup can also have a `) PROC` section and can request the user to provide input to answer questions.

So be creative with popups but not to overwhelm the user.

Dynamically Turning Off PFSHOW

PNPOP and RXPOPKEY

The RXPOPKEY example demonstrates how to disable and then re-enable PFSHOW if the user has PFSHOW ON. PFSHOW ON causes the function keys to be displayed at the bottom of ISPF panels. This is not always helpful in small popup panels.

```
/* ----- REXX ----- */
| This code demonstrates how to dynamically turn off, |
| and then back on, the PFSHOW setting. This is useful |
| if the user may have PFSHOW ON to prevent the function |
| keys from overlaying the popup. |
* ----- */
Address ISPEXec
call pfshow 'off'          /* make sure pfshow is off */
'addpop'
'display panel(testpop)'
'rempop'
call pfshow 'reset'      /* restore pfshow setting */
exit
/* ----- */
| The pfshow routine will: |
| 1. check to see the passed option |
| 2. if Off then it will save the current pfshow setting |
|    - save the current setting |
|    - turn off pfshow |
| 3. if the option is Reset then it will |
|    - test if pfshow was on and turn it back on |
* ----- */
pfshow:
arg pfkopt
if pfkopt = 'RESET' then do
  if pfkeys = 'ON' then
    'select pgm(ispopf) parm(FKA,ON)'
  end
if pfkopt = 'OFF' then do
  'vget (zpfshow)'
  pfkeys = zpfshow
  'select pgm(ispopf) parm(FKA,OFF)'
end
return
```

Progress Popup Panels

PNPROG1, PNPROG2, RXPROG1 and RXPROG2

There are times when an ISPF dialog needs to process behind the scenes and yet you want to let the user know that something is happening. This can be accomplished using a progress popup.

Here is a very simple example. The popup simply displays a counter, from 1 to 5, while the driver exec increments the counter and then sleeps for 1 second (simulating some background action).

Sample REXX code to demonstrate the Popup Panel:

```
/* rexx */
Address ISPEXec
do i = 1 to 5
  'control display lock'          /* Lock the Display */
  'addpop'                       /* Setup for a Pop Up Panel */
  pcount = i                     /* Update the variable for the panel */
  'display panel(pnprog1)'       /* Display the ISPF Panel */
  'rempop'                       /* Remove the Pop Up Setup */
  address syscall 'sleep 1'      /* Now sleep for 1 second to
                                simulate work. */
end
```

Here is the sample popup panel (named as PNPROG1):

```
)Attr
 @ Type(Output) intens(High)
)Body Window(24,3)
+
+Progress count:@z +
+
)Init
 &zwinttl = 'Progress Counter'
 .zvars = '(pcount)'
)Proc
)End
```

The output field, pcount, is defined using zvars as the field is only 3 characters on the panel. The use of zvars is not required here as the progress field can be anything – it could be a member name, a data set name, or some other value that will be helpful for the user.

Another option is a progress meter showing the progress of something where you can provide a percentage. This 'meter' adds two *'s for each increment of 10%:

```

/* rexx */
/* ----- *
| Provide a total number of iterations for the Demo |
* ----- */
arg total
if total = '' then total = 100
/* ----- *
| Change to ISPExec environment |
* ----- */
Address ISPExec
/* ----- *
| Define our increment so it fits in the panel |
* ----- */
incr = (total % 10) + 1
/* ----- *
| Define our increment indicator |
* ----- */
progc = '***'
i = 0
perc# = 0
/* ----- *
| Now loop thru and show the progress meter |
* ----- */
do until i = total
  i = i + 1
  if i//incr = 0 then do
    progc = progc'***'
    perc# = perc# + 10
    perc = perc#"%"
    prog = progc ('perc')
  /* ----- *
| Lock the display since the user will not |
| be able to do anything other than watch. |
| Then sleep for 1 second to simulate work. |
* ----- */
    "Control Display Lock"
    'addpop'
    'display panel(pnprog2)'
    Address syscall 'sleep 1'
    'rempop'
  end
end
end

```

And here is the Panel used for the Progress meter with a single output field that is updated with the 'meter':

```

)Attr Default(%+_ )
  @ type(output) intens(high) caps(off) color(blue)
)Body window(50,3) expand(\\)
+
+   Progress:@prog
+
)Init
  &zwinttl = 'Progress Meter'
)Proc
)End

```

Action Bars and Pull-downs

PNABC and RXABC

An action bar is the panel element located at the top of an application panel that contains action bar choices for the panel. Each action bar choice represents a group of related choices that appear in the pull-down associated with the action bar choice. When the user selects an action bar choice, the associated pull-down appears directly below the action bar choice. Pull-downs contain choices that, when selected by the user, perform actions that apply to the contents of the panel.

An action bar is analogous to the menu bar in Windows and macOS.

A panel can define pulldowns using the)ABC,)ABCINIT, and)ABCPROC panel sections.

- The)ABC section defines an action bar choice for a panel and its associated pull-down choices.
- The)ABCINIT section runs when the user selects that action bar choice.
- The)ABCPROC section runs when the user completes interaction with the pull-down choice and is optional.

```
)ATTR DEFAULT(%+_)
$ TYPE(AB) /* Action bar */
@ TYPE(ABSL) GE(ON) /* Action bar separator line */
)ABC DESC(Menu) MNEM(1)
PDC DESC('Save') ACTION RUN(SAVE)
PDC DESC('End') ACTION RUN(END)
PDC DESC('Cancel') ACTION RUN(CANCEL)
)ABCINIT
.ZVARS = 'MENUX'
)ABC DESC(Help) MNEM(1)
PDC DESC('Extended Help...')
ACTION RUN(XHELP)
)ABCINIT
.ZVARS = HELPX
)BODY WINDOW(48,6)
+$ Menu $ Help +
@-----+
% Action bars and pull-downs
%===>_ZCMD +
Action bars are cool. Right?_ans+ (Yes, No)
)END
```

And here is the sample display:

```
Compilers Test Help File Edit Edit_Settings Menu Utilities
- +-----+
-
E | Menu Help | Columns 00001 00072
C | +-----+ | Scroll ==> CSR
0 | | 1. Save | ars and pull-downs | +
0 | | 2. End | |
0 | | 3. Cancel | |
0 | +-----+ ool. Right? (Yes, No) |
* +-----+
*****
```


Verify a Data Set Name within Panel REXX

PNVDSN and RXVDSN

This example demonstrates how to use TSO services and how to set messages within a Panel.

The REXX code to display the panel is:

```
/* rexx */
address ispexec
do forever
  'addpop'
  'display panel(pnvdsn)'
  drc = rc
  'rempop'
  if drc > 0 then leave
end
```

Here is the ISPF Panel code:

```
)Attr DEFAULT(%+_ )
$ type(input) intens(low) hilite(uscore) caps(on)
)Body expand(\\) window(54,5)
+Enter/Verify Data Set Name:
%===>$verdsn          +
+
+Enter any data set name and press enter to verify.
                    +F3 to Exit.
)INIT
&zwinttl = 'Verify DSN Exists via Panel REXX'
)PROC
*REXX(* zedsmsg zedlmsg)
if sysdsn(verdsn) = 'OK' then do
  zedsmsg = 'Confirmed'
  zedlmsg = verdsn 'does exist.'
end
else do
  zedsmsg = 'Failure'
  zedlmsg = verdsn sysdsn(verdsn)
end
*ENDREXX
ver (&verdsn,nb,dsnameq)
.MSG = ISRZ001
)END
```

What this code does in the *REXX routine is to use the SYSDSN function to test the provided data set and then set the short and long message. Note the short and long messages are defined in the *REXX statement as they are not included within the panel body.

Then outside the REXX code there is the traditional ISPF Panel VER statement for the data set name and then the statement that generates the ISPF message.

Panel Scrolling Fields

PNSCRL and RXSCRL

Occasionally there are fields on a panel that are just too short for the data that is to be displayed, or entered, and the default fields will only truncate and not scroll. Fortunately, the option to enable field scrolling is easy.

Here is the sample Panel with a scrolling field:

```
)Attr Default(%+_)
/* ISPF - Developers - Tips and Tricks */
$ Type(input) hilite(uscore) caps(off) just(left) intens(high)
@ type(output)
)Body window(70,10) expand(\)
+--\-%Sample Scrolling Field Panel+\--\
%Command ==>_zcmd
+
+Enter data into the scrolling field and then scroll Right (F11) or
+Left (F10) to continue to enter data and to view it.
+
%=>$scrfld                                     @z
  @lrlsc
+   (Scroll Left or Right for up to 255 characters)
+
)Init
.cursor = scrfld
.zvars = '(scind)'
)Proc
)Field
Field(scrfld) ind(scind,'<>') len(255) scale(lrlsc)
)End
```

Let's look at the details of enabling a scrolling field:

1. Define the `)FIELD` section
2. Define each scrolling field using the `FIELD(field-name)` keyword and value
3. The `IND` keyword defines the scrolling indicator variable and characters. Using `<>` will use `<` to scroll left and `>` for scrolling right.
4. The `LEN` keyword defines the length of the scrolling field (255 in this case)
5. `SCALE` keyword defines a variable that will contain a scrolling ruler to help the user know the position within the field that is being displayed

This is what it will look like:

```
----- Sample Scrolling Field Panel -----
Command ==>

Enter data into the scrolling field and then scroll Right (F11) or
Left (F10) to continue to enter data and to view it.

==>
  ----+----1----+----2----+----3----+----4----+----5----+----6----+
  (Scroll Left or Right for up to 255 characters)
```

And this is the REXX code that can be used to experiment with the above panel:

```
/*  REXX  */
Address ISPEXec
do forever
  'addpop'
  'display panel(pnscr1)'
  xrc = rc
  'rempop'
  if xrc > 0 then leave
end
```

Dynamic Areas

PNDYN and RXDYN

This panel defines a dynamic area, but it also contains static text (field prompts), an output field (size), and an input field (somevar). All of these elements, and many more if needed, can coexist together.

The dynamic area defined in the panel is defined with `SCROLL (ON)` and `EXTEND (ON)`. This allows the application to scroll the data within the dynamic area when the user requests it, and ISPF will, at display time, extend the dynamic area depth so that the depth of the `)BODY` section equals that of the terminal that the panel is being displayed on.

It also shows how to use a shadow variable and character-level attributes to highlight individual characters within the string, overriding the field attribute for just one character, with no intervening space from a field attribute. The shadow variable is optional. Shadow variables are only needed when you want character level attributes.

```
)ATTR
@ AREA (DYNAMIC)          SCROLL (ON)  EXTEND (ON)
01 TYPE (DATAOUT)        COLOR (RED)
02 TYPE (DATAOUT)        COLOR (BLUE)
03 TYPE (DATAOUT)        COLOR (GREEN)
04 TYPE (DATAOUT)        COLOR (WHITE)
r TYPE (CHAR)  COLOR (RED)  HILITE (REVERSE)
g TYPE (CHAR)  COLOR (GREEN) HILITE (REVERSE)
b TYPE (CHAR)  COLOR (BLUE)  HILITE (REVERSE)
$ TYPE (TEXT)          COLOR (YELLOW)
)BODY
%----- EXAMPLE FOR USING A DYNAMIC AREA -----%
%COMMAND ===>_ZCMD                                     %SCROLL ===>_AMT +
%
+ This area is fixed.  size: &size
+
+ This is an input field%===>_somevar +
+
+This is extendable @DYNAREA,DYNSHAD @

$This should be at the bottom of the screen when in full screen.
)END
```

This REXX routine shows how to set up a variable for use as a dynamic area variable, how to set up the shadow variable associated with that area, how to display the panel, and how to process UP and DOWN scrolling. RIGHT/LEFT scrolling is ignored in this example.

The data to be shown is set up to imbed the hex characters '01'x, '02'x, '03'x, and '04'x to represent colors red, blue, green, and white respectively. It will highlight each of the color names in their respective color, and show a list with colored color names, and line numbers.

The exec also sets up a shadow variable, which highlights the first character of each line with a color in reverse video. This 'shadowing' effect is only available in dynamic areas and allows a change of the attribute at the character level, with no intervening 'blank' space caused by a field attribute.

Note that the shadow variable (variables shadata and dynshad in this example) is optional. You could leave it out of the panel and the exec, and you would not have character level coloring.

The display is handled by setting a variable which contains the program data (dyndata) and then setting a second variable which will display the data (DYNAREA). The second variable is needed in this case to handle scrolling. The same manipulation with the length of the dyndata variable is done to the associated shadow variable, dynshad.

The variable dyndata starts with the first line to be displayed. When a scroll request is received, the program calculates the offset into DYNDATA of the first displayed line and creates DYNAREA from there.

To display the dynamic area panel, use the following code:

```

/* REXX - A Dynamic area example */
Address ISPEXEC                                /* Calls go to ISPF      */
red   = '01'x                                  /* Assign colors to     */
blue  = '02'x                                  /* Attribute bytes     */
green = '03'x                                  /* found in the data   */
white = '04'x
maxlines = 600                                /* set max number of lines*/
dyndata = ''                                  /* initialize data      */
shadata = ''                                  /* initialize shadow var */

Do a = 1 to maxlines by 3                      /* Create some dummy data */
dyndata=dyndata||white||left('This is'red ||'red 'white||a , 29)
dyndata=dyndata||white||left('This is'blue ||'blue 'white||a+1, 29)
dyndata=dyndata||white||left('This is'green||'green'white||a+2, 29)
shadata=shadata||' r                          '
shadata=shadata||' b                          '
shadata=shadata||' g                          '
End

/* Add a bottom of data maker to the end of the data */
dyndata = dyndata||blue||centre(green||'BOTTOM' ||blue,29, '*')
shadata = shadata||' '

curline = 1;                                  /* set current line #   */

/* -----*/
/* Display loop until end or error */
/* -----*/

Do Until disprc > 0
dynarea = substr(dyndata,1+(curline-1)*30) /* set dynamic variable */
dynshad = substr(shadata,1+(curline-1)*30) /* set shadow variable */
size = length(dynarea)                    /* Set a scalar variable */
'ISPEXEC DISPLAY PANEL(PNDYN)'            /* Display the data     */
disprc = rc                                /* save return code     */
'ISPEXEC VGET (ZVERB,ZSCROLLA,ZSCROLLN)' /* get scroll values     */
Select                                     /* Process scrolling    */
  When(zverb = 'UP') Then                  /* Scroll up           */
    If zscrolla = 'MAX' Then              /* if scroll was max   */
      curline = 1                          /* scroll to top       */
    Else                                   /* else a number is known */
      curline = max(1,curline-zscrolln); /* (maximum is top)   */
  When(zverb = 'DOWN') Then               /* Scroll down        */
    If zscrolla = 'MAX' Then              /* if scroll was max   */
      curline = maxlines                  /* scroll to bottom    */
    Else                                   /* else a number is known */
      curline = min(maxlines,curline+zscrolln); /* (max is bottom)   */
  Otherwise;                               /* could use left & right too */
End
End                                          /* End of display loop */

```

This is what the dynamic area panel looks like when it is displayed.

```
----- EXAMPLE FOR USING A DYNAMIC AREA -----
COMMAND ==>                                     SCROLL ==> CSR
This area is fixed.  size: 18030
This is an input field ==>

This is extendable  This is red 1
This is extendable  This is blue 2
This is extendable  This is green 3
This is extendable  This is red 4
This is extendable  This is blue 5
This is extendable  This is green 6
This is extendable  This is red 7
This is extendable  This is blue 8
This is extendable  This is green 9
This is extendable  This is red 10
This is extendable  This is blue 11
This is extendable  This is green 12
This is extendable  This is red 13
This is extendable  This is blue 14
This is extendable  This is green 15

This should be at the bottom of the screen when in full screen.
```

Scrolling notes:

UP/DOWN scrolling

If all your data is in one variable, use (ZSCROLLN*area width) to locate the start of the displayed variable.

LEFT/RIGHT scrolling

Left/right scrolling usually involves creating a new variable to display. This is because the dynamic area uses a contiguous string.

When the user requests a scrolling action for the dynamic area, ISPF returns several variables which contain information about the requested scroll. This information is checked by the application and is then used by the application to reset the dynamic area variable.

ZVERB	Direction of scroll (UP, DOWN, LEFT, RIGHT)
ZSCROLLA	Amount to scroll (MAX, CSR, HALF, DATA, number, etc.)
ZSCROLLN	Number of lines to scroll

Additional information:

ISPF has several functions which can be used to help get additional information to assist in the definition of the dynamic area variable or scrolling requests.

PQUERY can help get information about areas contained within the panel. You specify the area name, and ISPF can return such things as the area type, and its width and depth.

The LVLIN built-in function can be called from the)INIT,)REINIT, or)PROC sections. It returns the line number of the last line within a dynamic, graphic, or scrollable area which was visible to the end-user on the currently displayed panel. This is extremely useful when the user can be in split-screen mode.

Your application can be coded to take advantage of larger-size terminal screens by using extendable areas, and by coding variable widths (on the)BODY section) and using the automatic expansion feature of ISPF. On the)BODY section, you code the EXPAND keyword, and then specify the characters you will place in the text of the panel when you wish to expand. The character in the)BODY that you place between these two characters will be used when the expansion is done.

ISPF Skeletons

ISPF skeleton definitions are stored in a skeleton library and accessed through the ISPF file-tailoring services. You create or change skeletons by editing directly into the skeleton library. ISPF interprets the skeletons during execution. No compilation or preprocessing step is required.

There are two types of records that can appear in the skeleton file:

- Data records - A continuous stream of intermixed text, variables, and control characters that are processed to create an output record.
- Control statements - Control the file-tailoring process.

The available control statements are:

) BLANK) CM) DEFAULT
) DO) DOT) ELSE
) ENDDO) ENDDOT) ENDREXX
) ENDSEL) IF) IM
) ITERATE) LEAVE) NOP
) REXX) SEL) SET
) SETF) TB) TBA

The file-tailoring services, listed in the order they are normally invoked, are:

- FTOPEN** Prepares the file-tailoring process and specifies whether the temporary file is to be used for output
- FTINCL** Specifies the skeleton to be used and starts the tailoring process
- FTCLOSE** Ends the file-tailoring process
- FTERASE** Erases an output file created by file tailoring.

File-tailoring services read skeleton files and write tailored output that can be used to drive other functions. Frequently, file tailoring is used to generate job control language statements for batch execution but can be used to generate many different outputs. The ability to code) REXX statements in the skeleton adds flexibility and power.

If output from file tailoring is not to be placed in a temporary file, the desired output file must be allocated to the ddname ISPFIL before invoking this service.

Simple Skeleton

RXSKLCMD and SKCMDS

Here is an example Skeleton used to generate a report on an ISPF Commands Table:

```
1. )TB 20
2. )SET CNT = 50
3. )DOT &TABLE
4. )SET CNT = &CNT + 3
5. )SEL &CNT > 50
6. 1 ISPF Command Table &TABLE          Date: &zdate Time: &ztime
7. )SET CNT = 1
8. )ENDSEL
9. 0CMD:  &ZCTVERB!&ZCTTRUNC
10. Action: &ZCTACT
11. Desc:  &ZCTDESC
12.)ENDDOT
```

In this skeleton the statements are:

1.)TB defines a tab which is used by the ! character to space data on the output record
2. Sets the counter to 50 to start with
3. Processed the table name in the variable &TABLE
4. Increments the counter by 3 for each table entry
5. Selects the next records to the)ENDSEL only if the counter is greater than 50
6. Inserts a record with a new page carriage control and a title
7. Resets the counter to 1
8. Ends the title record processing
9. Inserts 3 records into the output with information on the ISPF command table entry
12. Ends table processing

And here is the sample REXX code to process the skeleton:

```
/* ----- REXX ----- *
| Demonstration of ISPF Skeleton Processing |
* ----- */
Arg Table
if table = '' then table = 'ISPCMDS'
Address ISPEXEC
'ftopen temp'
'ftincl skcmds'
'ftclose'
'vget (ztempf)'
"View dataset('ztempf')"
```

In this Code, the parameter passed is the name of the table to process, and if blank use ISPCMDS. A temporary data set is used for the location of the generated report as indicated on the FTOPEN by using the TEMP option. After the File Tailoring is closed (FTCLOSE) a VGET for the variable with the name of the temporary data set is performed and that data set is then processed using View.

Skeleton with REXX

Using the TSO Stack

RXSKLRX and SKREXX

This example demonstrates using REXX and TSO services with a Skeleton.

This is the REXX code to drive the sample:

```
/* ----- REXX ----- *
| Demonstration of ISPF Skeleton Processing |
* ----- */
Address ISPEXec
'ftopen temp'
'ftincl skrexx'
'ftclose'
'vget (ztempf)'
"View dataset('"ztempf"')"
```

And here is the sample Skeleton which places some data into the TSO Stack and then pulls the data from the stack while generating JCL.

```
)CM Demo use of TSO stack in an ISPF skeleton
)CM Make list
)REXX STACKN
"delstack"
queue 'Kilroy'
queue 'was'
queue 'here'
stackn = queued()
)ENDREXX
)CM Make fixed front
//SKEL1 JOB (1), 'BACKUP', CLASS=A, COND=(0, LT), REGION=64M
//L EXEC PGM=IEBGENER
//SYSIN DD DUMMY
//SYSPRINT DD SYSOUT=*
//SYSUT1 DD *
)CM Make SYSUT1 from list in stack
)DO N = 1 TO &STACKN
)REXX DATA
parse pull data
)ENDREXX
&data
)ENDDO
)CM Make fixed back
//SYSUT2 DD SYSOUT=*
```

Using TSO Commands

```
RXSJKRXE abd SJREXXE
```

Note that in Skeleton REXX you can use most TSO services and commands, including ALLOC, EXECIO, and FREE. No ISPF services may be used.

Here is an example demonstrating the use of TSO commands and the TSO Queue:

```
)CM Demo use of TSO stack in an ISPF skeleton
)REXX STACKN
"alloc f(in) ds('sys1.proclib(bpxas)') shr reuse"
'execio * diskr in (finis stem in.'
'free f(in)'
"delstack"
  queue 'Reviewing BPXAS Proc'
  queue left('-',80,'-')
do i = 1 to in.0
  queue in.i
  end
  queue left('-',80,'-')
  queue ' '
  queue 'Including' in.0 'records.'
  stackn = queued()
)ENDREXX
)CM Get the data from the stack
)CM The variable STACKN is passed from the Skeleton REXX above
)CM with the number of records in the stack (queue)
)DO N = 1 TO &STACKN
)REXX DATA
  parse pull data
)ENDREXX
&data
)ENDDO
```

ISPF Tables

PNTAB and RXTAB

Tables are one of the capabilities of ISPF that are used frequently and yet rarely fully understood. There are so many nuances with the use of tables that are left to the developer. This section will attempt to clarify ISPF Tables.

Find

Find is different, in the opinion of the authors, than Locate (see [Locate](#) below). The author prefers to use Find to search for a string in any, or a subset, of the variables within a row.

Here is a code snippet from RXTAB for doing a FIND.

```
save_floc = 0    /* Save last find location */
save_find = null /* save last find string */
. . .
Tbdispl . . .
If zcmd = 'RFIND' then zcmd = 'FIND' save_find
Select
When abbrev("FIND",word(zcmd,1),1) = 1 then do
  find = translate(word(zcmd,2))
  save_find = find
  wrap_ = 0
  if save_floc > 0 then do
    'tbtop test'
    'tbskip test number('save_floc')'
  end
do forever
  'tbskip test'
  if rc > 0 then do
    'tbtop test'
    'tbskip test'
    if wrap = 1 then do
      zedsmg = 'Not Found'
      zedlmsg = find 'string not found in any member' ,
        'name. Try again.'
      'setmsg msg(isrz001)'
      leave
    end
  else wrap = 1
end
if pos(find,translate(vitem)) > 0 then do
  'tbquery test position(row)'
  crp = row
  save_floc = row
  if wrap = 1 then do
    zedsmg = 'Wrapped'
    zedlmsg = 'Find restarted at the top of the table.'
    'setmsg msg(isrz001)'
  end
  else do
    zedsmg = 'Found'
    zedlmsg = 'Found in row:' row
    'setmsg msg(isrz001)'
  end
  leave
end
end
end
end
```

This routine will search for the provided string in a single variable (vitem), however multiple row variables can be included in the `'if pos(' test.`

This code also detects when the end of the table is reached and will then 'wrap' to the top and start the find from there again, while issuing a message indicating that the find has 'wrapped'.

Enabling RFIND

Repeat Find, sadly, is not enabled by default with the out of the box ISPF. There are three ways to enable RFIND that the author has identified, each with their own requirements.

Update the ISPF Command Table

This method requires that an ISPF command table be updated with the following command:

Verb:	RFIND
Truncation:	0
Action:	&USRRFIND
Description:	User Repeat Find

This adds a command that has no action as the variable &USRRFIND will start out blank. The application must then set &USRRFIND to PASSTHRU to allow the RFIND command to be processed by the application, then reset it to blank when the application no longer needs to offer this command. A suggested approach is to enable before the use of TBDISPL and then after the TBDISPL completes, and before any actions are processed, then reset it.

Note that &USRRFIND is just an example and can be any variable name the application developer desires.

Within the code the developer must test if the command is RFIND and then change it to be processed to FIND followed by the FIND string

Create a Temporary Command Table for the Application

This method requires more coding but does not require that the sites ISPF administrator update any of the sites ISPF command tables.

Note that using this method requires a change if you use the REXX Compiler as `sysvar('sysicmd')` will return a null value when compiled. However, it appears that `sysvar('syspcmd')` does return the correct value when compiled.

For this to work the application code, very early, must do the following:

```
Address ISPExec
"VGET ZAPPLID"
if zapplid <> "TABT" then do
  "TBCreate tabtcmds names(zctverb zcttrunc zctact" ,
    "zctdesc) replace share nowrite"
  zctverb = "RFIND"
  zcttrunc = 0
  zctact = "&USRRFIND"
  zctdesc = "User controlled Repeat Find (RFIND)"
  "TBAdd tabtcmds"
  "Select CMD(%'sysvar('sysicmd') options ") Newappl(tabt)" ,
    "passlib scrname(TABLE)"
  Xrc = rc
  "TBClose tabtcmds"
  Exit xrc
end
```

What this does is:

1. Enter the ISPF environment
2. VGET the ISPF system variable ZAPPLID
3. Compare the ZAPPLID to our desired application ID of TABT
4. If they match, then we do nothing and all through to processing the real code
5. If they do not match, then
 - a. Create the temporary ISPF Commands Table. It must start with the application ID (TABT) and is defined with NOWRITE since there is no need to save it, with REPLACE in case it already exists and wasn't removed during a prior execution, and with SHARE to allow the table to be used across ISPF screens.
 - b. Define the four command table row variables:
 - i. zctverb is the command name
 - ii. zcttrunc is the number of characters to abbreviate the command (0 is no abbreviation)
 - iii. zctact is the action the command will initiate. For our purposes we specify an ISPF variable that we will update in the application.
 - iv. zctdesc is a description (can be blank)
 - c. Add that row to the command table
 - d. Reinvoke the active exec with a NEWAPPL which will enable the command table to be used.
 - e. Upon return from the exec the table is closed, and the exec Exits.

Locate a Row

Locate processing is less complex than a FIND as it is typically used to locate the 1st row with a string that matches the prefix of the selected value.

In this example the Locate always starts at the 1st row in the table. Remove the TBTOP to have the Locate process start at the currently displayed top of the table.

```
When abbrev("LOCATE",word(zcmd,1),1) = 1 then do
  vitem = translate(word(zcmd,2))
  'tbttop test'
  crp = 0
  'tbscan test arglist(vitem) position(crp) condlist(ge) '
  if rc = 0 then do
    zedsmg = 'Found'
    zedlmsg = word(zcmd,2) 'was found in row' crp
    'Setmsg msg(isrz001) '
  end
  else do
    zedsmg = 'Not Found'
    zedlmsg = word(zcmd,2) 'was not found'
    'Setmsg msg(isrz001) '
  end
end
```

In this example the code does the following:

1. Assuming this code is within a Select process the When tests for an abbreviation of the word LOCATE, allowing it to be abbreviated to a single character. Since the command would have multiple words, with the locate value being the second word, the abbreviation test must only work with the 1st word.
2. The row variable to be tested is then set to the translated to upper case 2nd word of the command.
3. The table is then reset to the top and the current row pointer set to 0.
4. Execute TBSCAN to scan through the table comparing the locate value to the value in each row. A successful match will place the row number in the variable crp.
5. The scan is using the CONDLIST option of GE, which means that the scan will stop when it finds a row that is greater than, or equal, to the locate value.
6. Based on the return code, a message is generated to the user that the value was found, or not found. When using a CONDLIST(GE).

Selecting Multiple Rows for Processing

Selecting multiple rows at one time requires a few additional lines of coding:

1. Set a variable (e.g. CRP) with the top displayed row of the table to 0 before the 1st TDBDISPL
2. Set the ZTDSELS variable to 0 before the 1st TDBDISPL
3. In the TDBDISPL code section
 - a. Test ZTDSELS and if 0 then
 - i. TBTOP
 - ii. TBSKIP to the save top display row (CRP)
 - iii. TDBDISPL with the PANEL parameter
 - b. If ZTDSELS is > 0 then
 - i. TDBDISPL without the PANEL parameter
 - c. Save the ZTDTOP value in a variable (e.g. CRP)

And that is it – the code will now support selecting multiple rows with one enter key.

Check out the full table processing code below to see how this all fits together.

Full Example

Using ISPF Tables can be challenging because it isn't obvious how to do some of the things that we are used to with the IBM, and other ISPF dialogs. But don't let that stop you – this chapter demonstrates some of the techniques that will make your ISPF table look polished and professional.

The first thing is to construct the ISPF Table display Panel. This panel was generated using the ISPF Edit Model command and selecting the TBDISPL Table Display Panel:

```
)Attr Default(%+_ )
/* % type(text ) intens(high) Defaults displayed for */
/* + type(text ) intens(low ) information only */
/* _ type( input) intens(high) caps(on ) just(left ) */
! type( input) intens(high) caps(on ) just(left ) hilite(uscore)
^ type(output) intens(low ) caps(off) just(asis )
)Body Expand(\)
%-\\- Sample Table Display Panel -\\-
%Command ==>_zcmd \ \%Scroll ==>_amt +
%
+Commands: %REF+refresh the display%Only string+Filter on string
+Selection:%S+Browse%SS/SS+Selection Range%S##+Selection Count
+
%Sel Status Item
+
)Model
!z + ^z + ^z +
)Init
.ZVARS = '(vsel vstate vitem)'
&amt = CSR
&vset = 0
)Reinit
)Proc
IF (&ZCMD = &Z)
if (&ztdsels = 0000)
&row = .csrrow
if (&row ^= 0)
if (&vsel = &Z)
&vsel = S
&vset = 1
if (&ztdsels ^= 0000)
&row = &Z
IF (&vsel ^= &Z)
if (&vsel = '=')
&vsel = &osel
&osel = &vsel
)End
```

This panel does several things:

1. Define the input field attribute of ! symbol to be used for the row selection with a hilite of uscore to show on the display with a _ to identify where to enter the selection command.
2. Defined an output field attribute using the @ symbol with caps off and justification of asis.
3. The scroll amount, which is always the 2nd input field on a table panel, is set to CSR (because the author prefers cursor scrolling instead of PAGE/HALF/##).
4. If the user has the cursor on a row and presses enter there is no selection, but this simulates a selection.
5. Note that IF statements are indentation sensitive.

Here is the REXX code, with comments, that displays the above panel.

```
/* ----- REXX ----- */
| Sample REXX Code to drive an ISPF Table to demonstrate |
| 1. Processing a command (Refresh) |
| 2. Processing individual line selections |
| 3. Processing multiple line selections |
| 4. Processing a range selection (SS/SS) |
| 5. Processing a range count (S###) |
| 6. Demonstrate Only string capability |
| 7. Uses IBM's ISRZ001 message (zedsmg/zedlmsg) |
| 8. Use of Find and RFIND |
| 9. Use of Locate (TBSCAN) |
* ----- */
```

The above is a short description of what the code is intended to do. This gives anyone reading the code a starting point to understand the process.

This next section does several things:

1. Establishes the REXX addressing environment to the ISPF (ISPEXec) environment.
2. Checks for a specific ISPF Application ID of TABT, and if that isn't the active APPLID then create a temporary ISPF Commands Table and add to it the Repeat Find command using a variable that will be updated later. Without this in the command table the Repeat Find command will never be passed to our code.
3. Next reinvokes the REXX code with the ISPF NEWAPPL keyword to place the code under the desired APPLID so that the command table will be in use.
4. Also set the screen name to Table for this screen.
5. Upon return then close the table and exit.

```
/* ----- *
| Define Addressing to ISPFExec (ISPF) |
* ----- */
Address ISPEXec
/* -----> Enable Repeat Find <----- *
| Check for APPLID (TABT) and if not then |
| - create an ISPF Commands Table |
| - add RFIND to it with a variable |
| - reinvoke the current exec with NEWAPPL |
| - on return close the commands table |
* ----- */
"VGET ZAPPLID"
if zapplid <> "TABT" then do
  "TBCreate tabtcmds names(zctverb zcttrunc zctact" ,
    "zctdesc) replace share nowrite"
  zctverb = "RFIND"
  zcttrunc = 0
  zctact = "&USRRFIND"
  zctdesc = "User controlled Repeat Find (RFIND)"
  "TBAdd tabtcmds"
  "Select CMD(%"sysvar('sysicmd') options ") Newappl(tabt)" ,
    "passlib scrname(TABLE)"
  x_rc = rc
  "TBClose tabtcmds"
  Exit x_rc
End
```

Next the code's defaults are defined. In this case only the null variable is defined.

```
/* ----- *
| Define defaults |
* ----- */
null    = '' /* Null variable for compare */
```

The Refresh label is used to restart processing at this location when the user invokes the Refresh command on the table.

```
/* ----- *
| Label for Refresh Processing |
* ----- */
Refresh:
/* ----- *
| Define our ISPF Table with two variables and NOWRITE as |
| it will not be saved.          |
* ----- */
```

At this point the table is created using TBCreate. For our purposes the table name is a literal with the name of test. This could easily be a variable so that the table name can be changed at will for various reasons.

The next set of code generates test data to fill the table with. This section can be replaced by your own code to populate the table or use an existing table.

```
"TBCreate test names(vitem vstate) nowrite"
/* ----- *
| Now fill the ISPF table with test data |
* ----- */
str = 'AbcDefGhiJklMnoPqrStuVwxYz0123456789'
strc = 1
vstate = null
do i = 1 to 100
  vitem = substr(str,strc,3) 'Test data number:' right(i+1000,3)
  strc = strc +3
  if strc > 34 then strc = 1
  "TBAdd test"
End
```

With the table populated it is time to define our defaults for processing. These variables are used to control some of the table processing actions.

```
/* ----- *
| Table defined and populated - get to the top now |
| and define our working variables          |
* ----- */
ztdsels = 0 /* Define # rows selected to 0 */
crp      = 1 /* Define the starting row to 1 */
save_floc = 0 /* Save last find location */
save_find = null /* save last find string */
```

A do forever loop is used to process the table until the user uses F3 (END) to terminate processing:

```
/* ----- *
| Process the table forever - until we say we're done |
* ----- */
do forever
```

Before the Table Display (TBDISPL) the usrrfind variable is changed to PASSTHRU and VPUT is used to update the variable in the ISPF pool. This allows the Repeat Find (RFIND) command to be passed to this code for processing. See [Enabling RFIND](#), above for more information.

```
/* ----- *
| Define USRRFIND for Passthru to enable RFIND |
* ----- */
usrrfind = 'PASSTHRU'
'vput (usrrfind)'
```

Next is the key to the table processing with the test of ztdsels (# of selected rows) so that we can process multiple rows at one time.

```
/* ----- *
| Test ZTDSELS - if greater than zero than display the |
| table without the Panel name |
| - if zero than display with the Panel name |
| after 1st resetting the display to the last location|
* ----- */
if ztdsels > 0
then 'tbd displ test'
else do
'tbtop test'
'tbskip test number('crp)''
'tbd displ test panel(pntab)''
End
```

The return code must be saved so that it can be tested but it shouldn't remain in the rc variable since the return code will be changed by the code that resets the USRRFIND variable using VPUT, which is done next to allow RFIND to be used by native ISPF routines (e.g. Browse, Edit, ...).

```
trc = rc /* Save the TBDISPL return code */
/* ----- *
| Reset USRRFIND to blank |
* ----- */
usrrfind = ''
'vput (usrrfind)'
```

At this point, after the USRRFIND has been reset, is the time to determine if the user used F3 to terminate processing and if so, then leave the do forever loop.

```
/* ----- *
| If the return code from the table display (TBDISPL) is |
| greater than 4 then we are done so leave the forever loop. |
* ----- */
if trc > 4 then leave
```

This is where the real processing occurs. There are both commands (zcmd) and row selections (vsel) and both are processed separately. For zcmd there is a need to check to see if the command is RFIND, something you'll never see if you didn't update the USRRFIND variable, and if it is RFIND then the zcmd must be changed to FIND followed by the last character string used for FIND. See [Find](#) for more information on supporting FIND, and [Enabling RFIND](#) for more information on RFIND.

```
/* ----- *
| Test the zcmd for any commands that we are to |
| process. |
* ----- */
if zcmd /= null then
if zcmd = 'RFIND' then do
zcmd = 'FIND' save_find
'tbtop test'
'tbskip test number('save_floc')'
end
Select
/* ----- *
| Find and Repeat Find processed. |
| |
| Checks for the provided string in the specified |
| row variables. |
* ----- */
When abbrev("FIND",word(zcmd,1),1) = 1 then do
find = translate(word(zcmd,2))
save_find = find
wrap = 0
if save_floc > 0 then do
'tbtop test'
'tbskip test number('save_floc')'
end
do forever
'tbskip test'
if rc > 0 then do
'tbtop test'
'tbskip test'
if wrap = 1 then do
zedsmg = 'Not Found'
zedlmsg = find 'string not found in any member' ,
'name. Try again.'
'setmsg msg(isrz001)'
leave
end
else wrap = 1
end
if pos(find,translate(vitem)) > 0 then do
'tbquery test position(row)'
crp = row
save_floc = row
if wrap = 1 then do
zedsmg = 'Wrapped'
zedlmsg = 'Find restarted at the top of the table.'
```

```

        'setmsg msg(isrz001) '
    end
    else do
        zedsmg = 'Found'
        zedlmsg = 'Found in row:' row
        'setmsg msg(isrz001) '
    end
    leave
end
end
end
end

```

This section of code demonstrates doing a Locate. See [Locate a Row](#) for more information on Locate processing.

```

/* ----- *
| Demonstrate Locate Processing on the row |
* ----- */
When abbrev("LOCATE",word(zcmd,1),1) = 1 then do
    vitem = translate(word(zcmd,2))
    'tbtop test'
    crp = 0
    'tbscan test arglist(vitem) position(crp) condlist(ge) '
    if rc = 0 then do
        zedsmg = 'Found'
        zedlmsg = word(zcmd,2) 'was found in row' crp
        'Setmsg msg(isrz001) '
    end
    else do
        zedsmg = 'Not Found'
        zedlmsg = word(zcmd,2) 'was not found'
        'Setmsg msg(isrz001) '
    end
end
end

```

The Only command (abbreviated to O) is used to limit the rows displayed to only those with the specified character string. This can be done a number of ways, but this is a simple technique, which requires using the Reset command to restore all the other rows to the table display.

```

/* ----- *
| Only will check for the provided string (case insensitive) |
| in any location within the row variables. |
* ----- */
When abbrev("ONLY",word(zcmd,1),1) = 1 then do
  str = subword(zcmd,2)
  if strip(str) = null then do
    zedsmg = 'Only Invalid'
    zedlmsg = 'Only requires a string parameter.'
    'Setmsg msg(isrz001) '
  end
  else do
    'tbttop test'
    do forever
      'tbskip test'
      if rc > 0 then do
        'tbttop test'
        leave
      end
      if pos(str,translate(vitem vstate)) = 0 then
        'tbdelete test'
      end
    end
  end
end
end
end

```

Refresh closes the active ISPF Table and then restarts the code at the label Refresh which recreates the table from scratch.

```

/* ----- *
| Refresh will close the table and then signal (goto) the |
| Refresh label to start over. |
* ----- */
When abbrev("REFRESH",zcmd,1) = 1 then do
  'tbend test'
  signal Refresh
end

```

Because no one is perfect there needs to be a way to let the user know when they enter a command that the code can't handle. This is one way to do that.

```

/* ----- *
| Process unknown commands here - basically let the user |
| know we didn't know what the command is that they |
| entered. |
* ----- */
Otherwise do
  zedsmg = 'Unknown'
  zedlmsg = zcmd 'is an unknown command - try again.'
  'Setmsg Msg(isrz001) '
end
end

```

After processing any command, it is time to check for any row selections. To keep the table from scrolling up, or down, this code saves the current top row (ztdtop) so that the table display can be repositioned when it is next displayed using TCDISPL.

```

/* ----- *
| Process the line selections if the selection variable (vsel) |
| is not null.          |
* ----- */
if strip(vsel) /= null then do

  /* ----- *
  | This code is to support the selection by the user |
  | pressing enter on the row and not entering any   |
  | selection value (point and shoot). |
  |          |
  | 1. save the current top row          |
  | 2. if the row selected via enter (vset = 1)      |
  |    then tbttop and then skip to that row      |
  * ----- */

  crp = ztdtop
  if vset = 1 then do
    vset = 0
    "TBTop test"
    "TBSkip test NUMBER("row")"
  end
end

```

This is where the individual row selections are processed. This example is very simple, but it demonstrates what you can do. There are also examples of handling multiple row selections and range selections. See [Selecting Multiple Rows for Processing](#) above for more information.

```

Select
  /* ----- *
  | Process an individual row |
  * ----- */
  When vsel = 'S' then do
    zedsmg = null
    zedlmg = 'Row data:' vitem
    'setmsg msg(isrz001)'
    say 'Row data:' vitem
    vstate = 'Selected'
    'tbput test'
  End

```

This next section of code demonstrates how to process a range of rows. See [Selecting Multiple Rows for Processing](#) for more information on this process.

```

/* ----- *
| Process a range selection where the 1st row is selected using SS |
| and then the last row with an SS.          |
* ----- */
When vsel = 'SS' then do
  /* ----- *
  | When only 1 SS is specified          |
  | if block indicator on then process it |
  | if not set the block indicator and   |
  | get the 1st row location |
  * ----- */
  if ztdsels = 1 then do

```

```

        if block = 1 then call do_ss
        else do
block = 1
vstate = 'SS'
'tbput test'
'tbquery test position(row1)'
row2 = 0
        end
        end
        /* ----- *
        | When Both from SS and to SS are provided |
        | - get the 1st row          |
        | - set row2 to 0|
        | - set block indicator to on |
        * ----- */
        else do
        'tbquery test position(row1)'
        row2 = 0
        block = 1
        end
        end
        /* ----- *
        | Process a selection with a count S###) |
        * ----- */
When left(vsel,1) = 'S' then do
        /* ----- *
        | Extract the count from the selection command |
        * ----- */
        vcount = substr(vsel,2)
        /* ----- *
        | Save current top row |
        * ----- */
        save_top = ztdtop
        /* ----- *
        | Validate that it is numeric |
        * ----- */
        if datatype(vcount) /= 'NUM' then do
        zedsmg = null
        zedlmsg = 'Invalid selection' vsel '- expecting S###'
        'setmsg msg(isrz001)'
        end
        else do
        /* ----- *
        | Get the current row number |
        | and the # of rows in table |
        * ----- */
        'tbquery test position(row1) rownum(rows)'
        /* ----- *
        | Identify the last row to process |
        * ----- */
        row2 = row1 + (vcount - 1)
        /* ----- *
        | Test for going past end of table |
        * ----- */
        if row2 > rows then do
zedsmg = null
        zedlmsg = vsel 'is invalid as it goes beyond the end of the table.'
'setmsg msg(isrz001)'
        end
        else do
        /* ----- *
        | Process the 1st row and update the state in the row |
        * ----- */
        say 'Row data:' vitem
        vstate = 'Selected'
        'tbput test'

```

```

/* ----- *
| Skip 1 row and process the rest |
* ----- */
'tbskip test'
do (row2-row1)
  'tbget test'
  say 'Row data:' vitem
  vstate = 'Selected'
  'tbput test'
  'tbskip test'
end
'tbtop test'
'tbskip test number('save_top')'
  end
  end
end

```

As for unknown commands, it is only polite to inform the user if they enter any row selection options that the code is unable to process.

```

/* ----- *
| Process unknown row selections |
* ----- */
Otherwise do
  zedsmg = 'Unknown'
  zedlmsg = vsel 'is an unknown row selection - try again.'
  'setmsg msg(isrz001)'
end
end
vsel = null
end
end

```

At this point the user has issued the F3 (END) command and processing is completed. The table must be closed, and the code can exit processing.

```

/* ----- *
| All done so close out the table |
| and Exit |
* ----- */
"TBEnd test"
Exit 0

```

Here is the subroutine used for range processing.:

```

/* ----- *
| Process the SS/SS Range command |
* ----- */
Do SS:
/* ----- *
| Get the row number of the 2nd SS |
* ----- */
'tbquery test position(row2)'
/* ----- *
| Test to see which SS entry is the 1st row to process and |
| arrange row1 to the lowest row number and row2 to the |
| highest row number. This allows the range to be selected |
| in either direction from any panel. |
* ----- */
if row2 < row1 then do
  rowx = row2
  row2 = row1

```

```

    row1 = rowx
end
/* ----- *
| Position to the top row and then skip down to the |
| 1st row to process.      |
* ----- */
'tbtop test'
'tbskip test number('row1')'
/* ----- *
| Process the selected rows |
* ----- */
do row2 = row1 + 1
  /* ----- *
  | Get the row data |
  * ----- */
  'tbget test'
  say 'Row data:' vitem
  /* ----- *
  | Update the row state variable |
  | Then update the row in the table |
  | Then skip to the next row.      |
  * ----- */
  vstate = 'Selected'
  'tbput test'
  'tbskip test'
end
/* ----- *
| When done then reset our working variables and return |
* ----- */
row1 = 0
row2 = 0
block = 0
ztdsels = 0
'tbtop test'
'tbskip test number('ztdtop')'
return

```

ISPF Messages

RXMSG

Every ISPF dialog needs to generate messages to inform the user of different situations. There are several ways to deal with messages, but the one trick makes it easy.

IBM provides a very useful, and generic, message in their ISRZ00 member – ISRZ002. This message is built using variables that anyone can use – so why create your own message when you can use this one.

To use this message, you need to:

1. Set variable ZERRSM to a short message
2. Set variable ZERRLM to a long message
3. Set variable ZERRALRM to YES or NO, to cause an audible alarm when the message is displayed
4. Set variable ZERRHM to the name of your tutorial panel
5. Use SETMSG MSG(ISRZ002) to generate the message

You can set the required ZERRALRM and ZERRHM variables early in your code and then change ZERRSM and ZERRLM multiple times depending on the circumstances.

Notes:

1. If ZERRSM is null, then the ZERRLM message will be immediately displayed.
2. If ZERRLM is null, then pressing F1 will display the tutorial panel instead of the displaying the long message
3. You can make a small popup for the long message by adding text in 75-character increments thus:

```
/* rexx */
Address ISPExec
zerrsm = 'Error'
zerralm = "NO"
zerrlm = left('test message',75,'*') ,
left('test message 2',75,'*')
zerrhm = "ispftutr"
"Setmsg msg(isrz002)"
```

Which generates this long message display:

```
+-----+
| test message***** |
| test message 2***** |
+-----+
```

Another option is to use the ISRZ001 message which uses variables ZEDSMMSG (short message) and ZEDLMSG (Long Message). For this you do not have to define alarm or help value. Keep in mind that the long message is limited to 512 bytes.

Note – if the short message value is null then the long message will be displayed without the user having to press F1. This is true for any ISPF message and not just the ISRZ001 or ISRZ002 messages.

ISPF Edit Macros

ISPF provides the ability to create your own ISPF Edit commands, called Macros, using different programming languages. As we have done throughout this document, we will use REXX, but that can be translated into Assembler, C, or your choice of language.

Symbolic Handling

This section will describe how to prevent an Edit Macro from interpreting an ampersand (&) as a variable when executing an Edit command.

The editor scans edit statements within program macros to do variable substitution like the CLIST processor. Only one level of substitution is done. This is the default; use the SCAN assignment statement to prevent it.

The SCAN macro command sets scan mode, which controls the automatic replacement of variables in command lines passed to the editor.

The SCAN assignment statement either sets the value of scan mode (for variable substitution) or retrieves the value of scan mode and places it in a variable.

1) Syntax

```
ADDRESS ISREDIT
  "(varname) = SCAN"
  "SCAN OFF"
  "SCAN ON"
```

2) Return Codes

0 Normal Completion

24 Severe Error

3) Example – No. 1

To set a line whose number is in variable &LNUM to:

&SYSDATE is a CLIST built-in function

Set scan mode off and issue the LINE command with &&SYSDATE as the CLIST function name. The CLIST processor strips off the first &, but, because scan mode is off, the editor does not remove the second &:

```
ISREDIT SCAN OFF
ISREDIT LINE &LNUM = "&&SYSDATE is a CLIST built-in function"
ISREDIT SCAN ON
```

Because the ISPEXEC call interface for REXX EXECs allows you to specify parameters as symbolic variables, a single scan always takes place before the syntax check of a statement. Therefore, the rule of using two ampersands (&) before variable names to avoid substitution of variable names also applies to REXX EXECs.

4) Example – No. 2

To exclude all lines and find/display all occurrences of "SYS1.SDITMOD1" with a volume of &SYSRS2, no matter how many spaces are between them:

```
"ISREDIT MACRO NOPROCESS"  
SPACE = ""  
VAR1 = "(SYS1.SDITMOD1)"  
VAR2 = "VOL(&&SYSR2)"  
LEN1 = LENGTH(VAR1)  
LEN2 = LENGTH(VAR2) - 1  
LEN3 = 71 - LEN1 - LEN2  
ADDRESS ISREDIT  
"SCAN OFF"  
"X ALL"  
DO J = 1 to LEN3  
  SPACE = SPACE" "  
  TEXT = VAR1||SPACE||VAR2          "F '"TEXT"' ALL"  
END
```

Centering Text using an Edit Macro

RXCENTER

This is an example of an ISPF Edit Macro that processes selected records by centering the data on them based upon either the data width or the width provided when the macro is invoked.

```
/* rexx */
Address ISREdit
'Macro (dw) NOPROCESS'
"PROCESS RANGE C"
"(start) = linenum .zfrange"
"(stop) = linenum .zlrange"
if dw = '' then
'(dw) = data_width'
do i = start to stop
'(data) = line' i
data = center(strip(data),dw)
'line' i '= (data)'
end
```

Copy this into your own REXX library as CENTER for use.

In this example:

1. The ISPF Edit addressing environment is defined.
2. The Macro statement identifies this code as an ISPF Edit Macro, with an optional parameter (dw) and instructing ISPF Edit to NOT Process (NOPROCESS) any line commands that may be found.
 - a. Note: NOPROCESS MUST be in UPPER CASE.
3. ISPF Edit is instructed to identify the RANGE of records selected, either with a single C for a single record, or CC on one record and CC on another to indicate a range.
 - a. Note that the range character(s) – C in this case must be UPPER CASE, the process range may be any case.
 - b. With PROCESS RANGE you can specify 1 or 2 characters to identify the range.
4. The range is then returned to the variables start and stop using the linenum service where zfrange (first) and zlrange (last) are the row numbers.
 - a. If no records were tagged with a C or CC/CC, then all records will be processed
 - b. C## can also be used to select a range
5. If the data width was passed as a parameter then use it, otherwise get the data width that ISPF Edit knows about.
6. Then loop from the start to the stop rows centering the data and updating the record using the LINE statement.

Invoking ISPF Edit with a Macro

RXEMAC and RXEMACE

It is very easy to call ISPF Edit from within code and have it open with a macro:

```
/* rexx */
dd = 'DD'random(99999)
'alloc f('dd') ds(temp.text) new spa(1,1) tr' ,
  'recfm(f b) lrecl(80) blksize(32720)'
macopt = 'This is a test of the ISPF Edit Macro Parm'
Address ISPEXec
'Control Display Save'
'Edit Dataset(temp.text) Macro(rxemac) Parm(macopt)'
'Control Display Restore'
Address TSO
'free f('dd') delete'
```

The 'control display save/restore' is very helpful to save the ISPF display state and restore it around the Edit.

In this example it:

1. Defines a random DDname to prevent conflicts with existing allocations
2. Allocates a data set
3. Defines the macopt variable which will be passed to the macro as a parm
4. Saves the display environment
5. Edits a data set with the data set name specifying the macro to use (rxemac) and the name of a REXX variable that contains the parm (macopt) to be passed to the Edit macro
6. Restores the display environment
7. Frees and Deletes the data set

Within the Edit Macro the parm is acquired from the macro statement:

```
/* rexx */
Address ISREdit
'macro (macopt)'
'caps off'
'number off'
'recovery on'
"line_after .zfirst = 'This is line number 1'"
"line_after .zfirst = 'This is line number 2'"
'line_before .zfirst = msgline (macopt)'
```

What this Edit Macro does is:

1. Define the ISPF Edit environment (Address ISREdit)
2. Indicate this is a macro and that a parm may be passed (macopt)
3. Turns Caps off, Numbers off, and turns Recovery on
4. Inserts 2 lines of text into the empty data set
5. Inserts a message with the passed parm text.

Passing Data to an Edit Macro

There are at least two easy techniques to pass data to an Edit Macro.

The 1st is demonstrated in the above section using the PARM keyword when invoking ISPF Edit. That is effectively the same as entering the data on the ISPF Edit command line when invoking the macro – see [Centering Text using an Edit Macro](#), above.

The 2nd technique is to have the Edit Macro use ISPF VGET services to retrieve the ISPF variables from the ISPF variable pool. This technique allows for more flexibility as more than a single value can be processed by using multiple ISPF variables.

Invoking an Edit Macro on All Members of a PDS

RXDOALL, RXPOPDO and RXPOPM

This example demonstrates how to invoke an ISPF Edit Macro on all members of a PDS. The macro can do anything but for this example the macro will change one value to another.

```
/* ----- rexx procedure ----- */
| Name:      RXDoAll      |
|           |
| Function:  This rexx exec will process the specified      |
|           ispf edit macro against every member of the    |
|           specified partitioned dataset.                  |
|           |
|           Only standard system services are used. The    |
|           LISTD TSO command with the MEMBERS keyword     |
|           is used to extract the member names.           |
|           |
| Syntax:    %DoAll dsname edit-macro |
|           |
| Sample Edit Macro to change SYS1 to SYS2                  |
| | rexx exec chsys1t2 (change sys1 to sys2) | |
| (the / was removed from the above line to avoid syntax  |
| errors in this exec).|
| Address ISREDIT      |
| "MACRO"              |
| "CHANGE 'DSN=SYS1.' 'DSN=SYS2.' ALL" |
| "SAVE"              |
| "END"                |
| Sample Execution:   %Doall 'sys2.testjcl' chsys1t2|
|           |
| ----- */

/* ----- */
| Get the target data set and macro |
* ----- */
arg dsn exec

/* ----- */
| Fix up data set name for use (no quotes) |
* ----- */
if left(dsn,1) <> "." then do
  dsn = sysvar(syspref)".dsn
end
else do
  dsn = substr(dsn,2,length(dsn)-2)
end

/* ----- */
| Setup Outtrap and do ListD |
* ----- */
x = outtrap("lm.", "*")
"LISTD" "'dsn'" "MEMBERS"
x = outtrap("off")

/* ----- */
| Process all members |
* ----- */
do i = 7 to lm.0
  parse value lm.i with mem extra
  Address ISPEXEC "EDIT DATASET('dsn('mem')') MACRO('exec') "
end
```

Another example of this is the RXPOPDO exec that will process all members of the target PDS and report on all ISPF Popup Panels, those members with)BODY and a WINDOW (parameter. The report will recommend a 'centered' AddPop location (row and column).

To use this exec execute it passing it the data set name of the PDS with the ISPF Panels. Here is a sample report for the example PDS:

```
ISPF Panel PopUp Centering Suggestions

Panel      ADDPOP Command
-----
$DEVCYPY  ADDPOP ROW(7) COLUMN(7)
$DEVVPX   ADDPOP ROW(7) COLUMN(6)
PNABC     ADDPOP ROW(7) COLUMN(17)
PNFLDH1   ADDPOP ROW(7) COLUMN(18)
PNFLDH2   ADDPOP ROW(7) COLUMN(18)
PNPOP     ADDPOP ROW(7) COLUMN(18)
PNPREXX   ADDPOP ROW(7) COLUMN(16)
PNPROG1   ADDPOP ROW(7) COLUMN(29)
PNPROG2   ADDPOP ROW(7) COLUMN(16)
PNSCRL    ADDPOP ROW(7) COLUMN(6)
PNVDSN    ADDPOP ROW(7) COLUMN(14)
```

The RXPOPM example is the ISPF Edit Macro which may be used by itself on any ISPF PopUp Panel and will insert a noteline with the recommendation:

```
***** ***** Top of Data *****
=NOTE= Suggested ADDPOP Command: ADDPOP Row(8) Column(18)
000001 )ATTR DEFAULT(%+ )
000002 /* ISPF - Developers - Tips and Tricks */
000003 @ type(output) caps(off) just(left)
000004 )BODY WINDOW(45,8)
000005 +
```

Changing ISPF Edit Commands with Macros

RXEM, RXEME, RXEMS and RXEMTRY

There are times when you need to change the behavior of an ISPF Edit command. This example demonstrates how to change the way both END and SAVE work by defining an Alias of those commands to an Edit Macro.

The RXEMTRY is a sample REXX exec that will drive ISPF Edit for this demonstration.

```
/* rexx */
Address ISPEXec
'vget (ztempf)'
"Edit dataset('ztempf') macro(rxem)"
```

RXEM is the primary ISPF Edit Macro that is called with the ISPF Edit command. It sets up the ISPF Edit environment by:

- Issuing a RESET ALL to remove a of the 'annoying' ISPF Edit information messages.
- Defining the ISPF End command to the RXEME Edit Macro.
- Defining the ISPF Save command the REXEMS Edit Macro.
- Turning off Caps and Numbers.
- Inserting 3 message lines to instruct the user how the demonstration works:

```

/* REXX */
Address ISREdit
'Macro'          /* First indicate we are an Edit Macro */
'Reset all'      /* Then Reset to turn off all messages */
'Define end alias rxeme' /* Now define END as an Alias */
'Define save alias rxems' /* And Save as an Alias */
'caps off'      /* Now turn off Caps */
'numbers off'   /* and turn Numbers off as well */

text = 'This demonstrates changing the Save and End commands.'
'line_after .zfirst = msgline (text) '
text = 'Enter the word DONE (any case) in a record to allow Save' ,
'or End to work.'
'line_after .zfirst = msgline (text) '
text = 'Or enter the CANCEL command to exit ISPF Edit.'
'line_after .zfirst = msgline (text) '

```

Both ISPF END and SAVE commands will not invoke the RXEME and RXEMS Rexx exec's which will both check for the word 'DONE', in upper/lower/mixed case, and only then process the real ISPF command. If the word is not found, then the user is informed using ISPF messages.

The RXEME code is below:

```

/* REXX */
Address ISREdit /* Setup Addressing to ISPF Edit */
'Macro'        /* Indicate we are a macro */
'Find 'done' first' /* Do a FIND for the word 'DONE' */
if rc = 0 then do/* if found then: */
  'define end reset' /* Reset the alias definition for END */
  'define save reset' /* Reset the alias definition for SAVE */
  'end' /* Do the real Edit END command */
  exit 0 /* And edit the macro */
end

zedsmg = 'Warning'
zedlmsg = left('To exit ISPF Edit enter the word DONE somewhere',75) ,
'in the text or enter the CANCEL command.'
Address ISPEXEC
'setmsg msg(isrz001)'

```

And the RXEMS code:

```

/* REXX */
Address ISREdit /* Setup Addressing to ISPF Edit */
'Macro'        /* Indicate we are a macro */
'Find 'done' first' /* Do a FIND for the word 'DONE' */
if rc = 0 then do/* if found then: */
  'SAVE' /* Do the real Edit SAVE command */
  exit 0 /* And edit the macro */
end

zedsmg = 'Warning'
zedlmsg = left('To Save the data enter the word DONE somewhere',75) ,
'in the text or enter the CANCEL command to exit.'
Address ISPEXEC
'setmsg msg(isrz001)'

```

Library Services

LMDLIST – List Data Sets

RXLMD

LMDLIST will return to the application information on all the data sets under the provided high-level-qualifier. This example demonstrates that in a simple way. Note that this code can also be executed in batch as there are no display panels used.

```
/* ----- rexx ----- *
| This REXX code demonstrates using the LMDLIST ISPF Service |
| to list information about data sets under the provided |
| high-level-qualifier (hlq). |
| |
| If no hlq is provided then the users prefix or userid |
| will be used. |
* ----- */
arg hlq
/* ----- *
| initialize our working variables |
* ----- */
parse value ' ' with null dsn
parse value '0 0 0' with total count used
/* ----- *
| Validate and/or set the HLQ |
* ----- */
if hlq = null then do
  if sysvar('syspref') /= null then
    hlq = sysvar('syspref')
  else hlq = sysvar('sysuid')
end
/* ----- *
| Define ISPF Environment |
* ----- */
Address ISPEXEC
/* ----- *
| Setup LMDINIT |
* ----- */
"Lmdinit Listid(LMD) Level("hlq")"
/* ----- *
| Process ALL data sets under the HLQ |
* ----- */
do forever
  "Lmdlist Listid("lmd") Stats(YES) Dataset(dsn) Option(LIST)"
/* ----- *
| If return > 0 then finish up |
* ----- */
  if rc > 0 then do
    "Lmdfree listid("lmd")"
    say "All done - count:" count 'used:' used
    exit 4
  end
/* ----- *
| Update counter and report out |
* ----- */
  count = count + 1
  say count:" " "Dataset:" dsn "Vol:" zdlvol "Used:" zdlsize ,
    "Device: *"zdldev"*" ,
    "Mig:" zdlmigr c2x(zdldev)
  sysvolume = ""
  if datatype(zdlsize) = 'NUM' then
    used = used + zdlsize
end
```

Sample Batch JCL to run this example:

```
//JOBNAME JOB (BATCH),REGION=0M,NOTIFY=&SYSUID,CLASS=A
//OUT OUTPUT DEFAULT=YES,JESDS=ALL,OUTDISP=(HOLD,HOLD)
//* ----- *
//* Demonstration of running an ISPF application in Batch. *
//* ----- *
//TSO EXEC PGM=IKJEFT01
//SYSEXEC DD DISP=SHR,DSN=hlq.DEVTIPS.PDS <=== Change
//SYSTSPRT DD SYSOUT=*
//SYSTSIN DD *
    ISPF CMD(%TESTLMD)
//ISPPROF DD UNIT=VIO,
//SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=6160)
//ISPMLIB DD DISP=SHR,DSN=ISP.SISPMENU
//ISPPLIB DD DISP=SHR,DSN=ISP.SISPPENU
//ISPPLIB DD DISP=SHR,DSN=ISP.SISPPENU
//ISPPLIB DD DISP=SHR,DSN=ISP.SISPSENU
//ISPTLIB DD UNIT=VIO,DISP=(NEW,PASS),SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=27920)
// DD DISP=SHR,DSN=ISP.SISPTENU
//ISPLOG DD DUMMY,DCB=(LRECL=120,BLKSIZE=2400,DSORG=PS,RECFM=FB)
```

Miscellaneous

Browsing Data in a REXX Stem

RXSTEM RXSTEME

Many times, an ISPF dialog will collect information in a REXX STEM variable and need to present that to the user. This approach is very simple using a temporary data set and ISPF Library Services.

```
/* ----- REXX ----- *
| Sample routine to browse data from a stem variable. |
* ----- */
do i = 1 to 200 /* Define our test data */
  stem.i = 'Test data record number' right(i+1000,3)
end
stem.0 = i /* update the stem.0 with record count */
/* ----- *
| Randomly define a DDName to use |
* ----- */
dd = 'dd'random(9999)
/* ----- *
| Allocate a temporary data set for our data |
* ----- */
'Alloc f('dd') new spa(5,5) tr' ,
'recfm(v b) lrecl(80) blksize(0)'
/* ----- *
| Write out the stem data |
* ----- */
'Execio * diskw' dd '(finis stem.)'
/* ----- *
| Access the Temporary Data Set using ISPF |
| Library Services. |
| Then using ISPF Browse service to browse the data. |
| And use Library Services to Free the Data Set. |
* ----- */
Address ISPEXec
'lmnit dataid(ddb) ddname('dd')'
'browse dataid('ddb')'
'lmfree dataid('ddb')'
/* ----- *
| Last Free the z/OS Allocation |
* ----- */
Address TSO
'Free f('dd')'
```

A better and easier approach is to use the STEMEDIT command which can be found on the CBTTape in File 895. Here is the code for that:

```
/* ----- REXX ----- *
| Sample routine to browse data from a stem variable. |
* ----- */
do i = 1 to 200 /* Define our test data */
  stem.i = 'Test data record number' right(i+1000,3)
end
stem.0 = i /* update the stem.0 with record count */
call stemedit 'Browse',stem.,,'Stem stem. Browsing'
```

ISPF in Batch

JCBAT1 and JCBAT2

For ISPF in batch the batch TMP (terminal monitor program) is used. It is recommended to use IKJEFT1B which has the advantage over IKJEFT01 of returning the return code of the last executed command.

Below is an example of running the IBM TASID program in batch. This is an ISPF application written by Doug Nadel.

The Snapshot utility (Option 8) allows you to create a sequential file which contains data obtained by most of the TASID options. The name of the data set can be changed by using the Settings pulldown on the TASID main menu.

The PRINTDS command is used in this example to obtain a hardcopy of data set userid.TASID.SNAPSHOT through foreground copying to SYSOUT.

```
//JOBNAME JOB (BATCH),REGION=0M,NOTIFY=&SYSUID,CLASS=A
//OUT OUTPUT DEFAULT=YES,JESDS=ALL,OUTDISP=(HOLD,HOLD)
//TSO EXEC PGM=IKJEFT1B
//STEPLIB DD DISP=SHR,DSN=TASID.LOAD <=== Change
//SYSTSPRT DD SYSOUT=*
//SYSTSIN DD *
PROFILE PREFIX(userid) <=== Change
ISPSTART PGM(TASID) PARM(8)
PRINTDS DSNNAME(TASID.SNAPSHOT) CCHAR NOTITLE SYSOUT(R) HOLD
//ISPPROF DD UNIT=VIO,SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=6160)
//ISPMLIB DD DISP=SHR,DSN=ISP.SISPMENU
//ISPLIB DD DISP=SHR,DSN=ISP.SISPPENU
//ISPSLIB DD DISP=SHR,DSN=ISP.SISPSENU
//ISPTLIB DD UNIT=VIO,DISP=(NEW,PASS),SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=27920)
// DD DISP=SHR,DSN=ISP.SISPTENU
//ISPLOG DD DUMMY,DCB=(LRECL=120,BLKSIZE=2400,DSORG=PS,RECFM=FB)
```

This is another example:

```
//JOBNAME JOB (BATCH),REGION=0M,NOTIFY=&SYSUID,CLASS=A
//OUT OUTPUT DEFAULT=YES,JESDS=ALL,OUTDISP=(HOLD,HOLD)
//TSO EXEC PGM=IKJEFT1B
//SYSEXEC DD DISP=SHR,DSN=hlq.DEVTIPS.PDS <=== Change
//SYSTSPRT DD SYSOUT=*
//SYSTSIN DD *
ISPF CMD(%TESTLMD)
//ISPMLIB DD DISP=SHR,DSN=ISP.SISPMENU
//ISPTLIB DD DISP=SHR,DSN=ISP.SISPTENU
//ISPPROF DD UNIT=VIO,SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=6160)
//ISPLIB DD UNIT=VIO,DISP=(NEW,PASS),SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=27920)
//ISPSLIB DD UNIT=VIO,DISP=(NEW,PASS),SPACE=(TRK,(1,1,5)),
//DCB=(RECFM=FB,LRECL=80,BLKSIZE=27920)
//ISPLOG DD DUMMY,DCB=(LRECL=120,BLKSIZE=2400,DSORG=PS,RECFM=FB)
```

The only 'real' data sets required in batch is ISPMLIB and ISPTLIB, the other ISPF data sets can be temporary.

Randomize DDname and Table Names

RXRAND

There are several techniques for randomizing both DDnames and ISPF Table names with a goal to make sure that the names are unique.

Based on the REXX Exec Name

This technique uses the REXX PARSE SOURCE command to get the name of the REXX Exec that is being executed and then use that for the DDname or a temporary Table name. This information can also be used for work data set names that are allocated using a data set name. One advantage to this approach is that it is easier when debugging to determine the REXX code 'causing' the 'problem'.

To get the active REXX Exec name use this sample code:

```
/* rexx */
parse source . . exec ddname .
say 'exec:' exec
say 'ddname:' ddname
```

And what you get back is this when the code is named TSOURCE:

```
exec: TSOURCE
ddname: SYSEXEC
```

Using the REXX Random Function

This technique is more flexible while reducing the debugging information;

```
/* rexx */
dd = 'DDN'random(99999)
say 'ddname:' dd
```

With this approach the variable DD is assigned a value that starts with DDN followed by 5 random numbers between 1 and 99999. The character string DDN can easily be changed to something meaningful that will help in debugging.

Another advantage is that if this is used in an ISPF dialog that is opened in multiple ISPF screens, then you can guarantee that there will be no DD name or Table name conflicts.

Another Random String

Another approach is to use code similar to this:

```
/* ----- *
| Using the day of month and second since midnight |
* ----- */
ddn = 'D'left(date('e'),2)''right(time('l'),5)
say 'DDname:' ddn
```

With this approach we use the 2 digit day of the month and the 5 digits for the seconds since midnight, all preceded by a character to yield an 8 character DDname.

Getting the DDname from the System

In this example, using BPXWDYN, the ddname is determined by the system and returned into a variable (dd1). Then the data in the data set is read using execio, and the allocation then freed using BPXWDYN.

This code requires that the user have OMVS access.

```
/* REXX */
ds='your.dataset.name'
cc=bpwdyn('alloc da('ds') shr rtdn(dd1)') /* return ddname in dd1 */
if cc<>0 then say alloc' ds 'rc' cc
else do
  "execio * diskr" dd1 "(stem data. finis)"
  cc=bpwdyn('free dd('dd1')')
end
```

Stem Sort

Consider the following directory list application. To facilitate sorting by various fields, a directory stem named `dir.` is used.

Name	GenNum	VV	MM	Created	Changed	Size	Init	ID
. \$\$\$\$\$\$	0	01.01	2019/04/20	2019/04/20	09:23	2	1	TRIDJK
. \$\$\$\$\$\$	1	01.06	2019/04/20	2019/05/03	12:27	5	1	TRIDJK

The REXX code snippet below shows how to set up a stem sort using BPXWUNIX SORT. Sort keys are defined for each field in the stem to be sorted. For the `CHANGED` and `SIZE` fields, a primary and secondary key are used. For the `CREATED` and `ID` fields, only a primary key is used.

Note that the input and output to the sort use the same stem.

```
created = '-dr -k5,5'
changed = '-dr -k6,6 -k7,7'
size = '-nr -k9,9 -k8,8'
id = '-dr -k11,11'

if abbrev("CREATED",word(column,1),2) = 1 then
  call do_sort created
if abbrev("CHANGED",word(column,1),2) = 1 then
  call do_sort changed
if abbrev("SIZE",word(column,1),1) = 1 then
  call do_sort size
if abbrev("ID",word(column,1),1) = 1 then
  call do_sort id

do_sort:
  parse arg key
  env.0=1
  env.1='LC_COLLATE=S370'
  stderr.0 = 0
  call BPXWUNIX "/bin/sort "key, ,
               "dir.", "dir.", "stderr." "env."
  return
```

Full Stem Sort example

RXSTEMS

This example is self-contained to sort a stem.

```
/* ----- REXX ----- *
| From John McKown to sort a stem using unix sort in OMVS |
* ----- */
XX=SYSCALLS('ON')
If xx > 3 Then Do
  Say "SYSCALLS('ON') Failed. RC="xx
  exit
End
do i = 1 to 12
  a.i = random(999)
end
a.0=12
stdout.0=0 ; stderr.0=0
call bpxwunix "/bin/sort -n",a.,a.,stderr.
say 'a.0='a.0
do i=1 to a.0
  say "a."i"="a.i
end
say 'stderr.0='stderr.0
do i=1 to stderr.0
  say "stderr."i"="stderr.i
end
```

First is a verification that the user has access to OMVS. Next a stem with 12 elements is created using random 3-digit numbers.

More on Stem Sorting using BPXWUNIX

'Borrowed from an IBM-Main posting by Albert Ferguson:

```
Under zOS REXX does have a STEM Sort (via BPXWUNIX). If you have an OMVS
Segment on your USERID (and as of zOS 2.1+ you should), try this:

/*
  Do a Binary Sort (use -tn for Text)
  Using 2nd Word in each StemIn. Record as primary sort key
  Using 1st Word in each StemIn. Record as secondary sort key
  More doc on Unix Sort
at http://publibz.boulder.ibm.com/epubs/pdf/bpx1cd10.pdf
*/
x = BPXWUNIX("/bin/sort -bn -k2,2 -k1,1","StemIn.,"StemOut.")

Also available via the USS side of zOS is Regex support:

/*
  Get the JESMSG LG output from a JOB, e.g. via SDSF or (E)JES and put in the
STEM JesMsgLg.
  Use the USS grep in, Extended mode, to search for all STEPS with an RC=12,
16, or 20
  Return only those steps to the BadSteps. STEM variable ...

  Add a step using this at the very END of a JOB to determine if something
did not go as expected, and then
possibly correct it!
*/
x = BPXWUNIX("/bin/grep -E -e'↵.{30}(.{10} (12|16|20) '",
"JesMsgLg.,"BadSteps.")
```

Converting the User Provided Data Set Name to a Full Data Set Name

Many times, the user, when prompted to enter a data set name, will enter the data set name without fully qualifying it, and without the default TSO Prefix. That isn't a problem as all references to the data set name are using standard system services that will resolve the data set name properly. If the data set name must be fully qualified within the application the easy way to do that is to use the TSO LISTDSI service, which will return the fully qualified data set name (without quotes) in the SYSDSNAME variable.

```
/* rexx */
arg dsname
x = listdsi(dsname)
Say 'Input DSN:' dsname
Say 'Full DSN:' sysdsname
```

Note that this works whether the user is running with, or without, a TSO PREFIX.

Default ISPF Terminal Type

It has been recommended that the ISPF Terminal Type 6 (3278T) model be selected from ISPF Primary Menu Option 0.

Useful Tools

IBM ISPF includes several useful tools, among these are:

ISRDDN

Use this tool before and after testing your application to verify that all allocations have been released. To use it issue TSO ISRDDN from any ISPF Command line.

ISPLIBD

This ISPF command will display all the current ISPF LIBDEF allocations. To use it issue ISPLIBD on any ISPF Command line. Note that this will only show the allocations for the current ISPF Screen and not on other split screens.

ALTLIB Display

The ALTLIB command, with the Display parameter, will display the current ALTLIB allocations. It only displays the DDnames and not the data set names, which is a shame.

Debugging Hints/Tips

Displaying the ISPF Panel Name

Many times, it is useful to know which ISPF Panel is being displayed. To display the active Panel name, issue the **PANELID ON** command and then the reverse is **PANELID OFF**.

ISPF Dialog Test

ISPF dialog test (Primary Option 7) is a dialog that can be used to test your ISPF applications.

```
----- Dialog Test -----
Menu  Utilities  View  Help
      Primary Option Panel
Option ===>

1 Functions      Invoke dialog functions/selection panel
2 Panels          Display panels
3 Variables       Display/set variable information
```

4	Tables	Display/modify table information
5	Log	Browse ISPF log
6	Dialog Services	Invoke dialog services
7	Traces	Specify trace definitions
8	Breakpoints	Specify breakpoint definitions

Whilst there are many options in Dialog Test, *Traces*, *Breakpoints* and *Functions* are the options most commonly used.

A typical debugging session begins by “activating” traces (functions and variables) and breakpoints. This is done by specifying YES in the “Active” column for these definitions.

After setting your definitions, start your dialog using *Functions* option of Dialog Test. When the dialog completes, you can browse the ISPF log (Option 5) to see the results. If your Log is allocated to SYSOUT, then you can use SDSF to browse the ISPLOG after accessing the Job Data Set Display panel.

REXX Trace

1. TRACE controls the tracing action, how much information is displayed, during processing of a REXX program. TRACE is mainly used for diagnostic purposes.
 - a. TRACE I (Intermediates) traces all clauses before execution and traces intermediate results during evaluation of expressions.
 - b. TRACE ?i will pause after each display, allowing the user to interactively issue any REXX command to assist with debugging.
 - c. TRACE OFF, or just TRACE, will turn off tracing.
2. Tracing can also be enabled from the ISPF Command line using the TSO commands:
 - a. TSO EXECUTIL TS to initiate tracing using the ?i option
 - b. TSO EXECUTIL TE to terminate tracing
3. Tracing can also be performed in both Panel REXX and Skeleton REXX.

Miscellaneous Tips

1. Use ISPVCALL before and after to trace general stuff
2. Use ISPFTRC to trace and capture skeletons before and after. Great tool. Requires DD for ISPFTRC to be allocated.

Example:

```
if debug          /* show all work to ISPFTRC DD          */
  then ADDRESS TSO "ISPFTRC READ(DETAIL) RECORDS(*) TBV(DETAIL)"
"FTINCL S@CPYHCD" /* RC=8 usually means member not found */
lrc = rc          /* RC=20 usually means variable error */
```

3. See Appendix C of ISPF Dialog Developers guide for notes on Panel and File Tailoring Traces
4. The use of)REXX section in skeletons to display variables. The calling REXX exec issues TSO “ISPFTRC xxxxx”.
5. if you get RC=20 on FTINCL, display zerrlm. This assumes “ISPEXEC CONTROL ERRORS” is set.

```
ADDRESS ISPEXEC "FTINCL `myskel`"
ftincl_RC = rc
if ftincl_RC <> 0
then do
say ZERRSM ZERRLM /* optional */
call error_rtn "FTINCL" ftincl_rc myskel ,
"ZERRSM(`zerrsm`) ZERRLM(`zerrlm`)"
```

```
end
```

6. Skeleton notes

7. Review operators, built-in functions, and control statements

```
)SEL &EML= Y  
is not valid .    this is valid  
)SEL &EML = Y
```

8. Be wary of special characters like "<" ">"

```
//* look for errors > 0    - will fail  
//* look for errors >>0   - will work
```

9. May need to utilize "SCAN OFF" and "SCAN ON"

10. Temporary datasets. Will need review for "&"

```
// DD DSN=&&&&&TEMP,      will resolve to  
// DD DSN=&&TEMP,
```

11. Watch that tailored line does not exceed the allowed record length.

Appendix

Useful Tools

These tools, along with the sample code for the various chapters, are included in a TSO Transmit (XMIT) format file that is shared with this document.

CMT

CMT

CMT is an ISPF edit macro to simplify entry of comments. CMT enters all comments in the syntax of the object being edited and formats them via sent/defaulted user parameters. Defaults for execution are set/inquired via the CMT syntax #2, shown below, and some may be overridden via the main CMT syntax #1.

All comments require a target line be defined in your data.

Syntax 1 to add comment line(s) (all parms optional)

CMT #,B,text

- #=INDENT 1 - 70, C (Center) or R (Right Justify) box
- B =BOXSIZE F S (Fixed 65 chars) or V (Variable size)
- cmt text blank shows popup panel to enter multiple lines.

Syntax 2 to Set CMT defaults (also may be set on popup panel)

CMT SET <parmId> <parm value>

parmId Indent, Boxsize, Charbox or Fixlength
parm value for I: #1-70, C or R B: F,S or V
C: any character F: 1-65

Download from <http://lbdsoftware.com/ispftools.html>.

LOADISPF/DROPISPF

LOADISPF

LOADISPF and DROPISPF are a pair of REXX routines to be copied into your REXX code that supports placing all ISPF elements (panels, messages, skeletons, CLISTs, and REXX) inline in your REXX exec. It will, when called, scan your source code and dynamically allocate temporary datasets and then ALTLIB or LIBDEF to them. Note that DROPISPF is included in the LOADISPF member since they go together.

This is very useful, as it allows the application to be completely self-contained. There is no need to distribute, or install, multiple elements when all of the REXX routines needed, all the ISPF Messages, Panels, and Skeletons can be installed with one REXX exec.

See in the provided sample PDS member TESTPSL, which is the Point-and-Shoot sample dialog contained within this single REXX exec. The ISPF Panel is found inline and uses the LOADISPF/DROPISPF routines to dynamically load and access the panel using LIBDEF.

Download both in the LOADISPF package from <http://lbdsoftware.com/ispftools.html>.

REXXFORM

#RXFORM

REXXFORM is an ISPF edit macro used to format REXX EXECs by indenting DO and SELECT groups and left-justifying lines at a selected column. REXXFORM also checks for unbalanced DO and SELECT statements while providing a consistent format to your REXX code that helps improve readability and maintainability.

REXXFORM was originally developed as a tool to format REXX code under XEDIT under z/VM. It was ported to z/OS as an ISPF Edit Macro with some enhancements to take advantage of the ISPF environment.

This command may only be used from the ISPF Edit command line.

Command Syntax is:

REXXFORM ?	Display ISPF Tutorial Panel
REXXFORM left-margin indent-column (options	Left-margin is the column all lines are justified to (default is 4) Indent-column is the column that DO and Select statements are indented to from the left-margin (default is 3) Options are: CI to justify comments to the left-margin CJ to justify comments to column 1

By default, all records will be processed.

To select an individual record or a range use the following syntax:

- S Select a single record to format
- SS/SS Select a range of records to format
- S# Select current record for # -1 (S4 is current plus 3)

Download from <http://lbdsoftware.com/ispftools.html>

STEMEDIT

This assembler sub-routine can be invoked by a REXX EXEC to display the contents of stem variables using the ISPF BRIF, VIIF or EDIT services. STEMEDIT is a nice complement to the REXX OUTTRAP function, when it is used in the ISPF/PDF environment.

Download File 895 from www.cbttape.org.

TRYIT

#TRYIT

Overview

TRYIT is an ISPF Edit command that is designed to be used to test an Assembler program, CLIST, REXX Exec, JCL, ISPF Panel or ISPF Skeleton while it is being edited. The way this works is such that the JCL, CLIST, REXX Exec, ISPF Panel, or ISPF Skeleton does *not* must be in a library in the existing SYSPROC, SYSEXEC, ISPPLIB, or ISPSLIB allocations thus allowing the development and testing in other, less critical, data sets.

If a JCL Syntax checking product is available, then TRYIT can be used to invoke it - this is assuming the product can be invoked as an ISPF Edit Macro (e.g. CA-JCLCheck and JCLPrep).

For Assembler programs the active member will be assembled and optionally linked into a specified target library. After entering TRYIT the user will be prompted to enter the assembly and linkedit information if the member is determined to be an assembler program.

For CLIST and REXX Exec members the active data set in which the member resides will be allocated using the TSO ALTLIB facility and then the member executed, along with any passed optional parameters.

For ISPF Panels and ISPF Skeletons the active data set in which the member resides will be allocated using the ISPF LIBDEF facility then then the panel Displayed or Selected based upon the parameters provided to TRYIT. If there are any errors in the panel or skeleton an ISPF message will be displayed and the error may then be corrected using ISPF Edit and TRYIT used once again to verify the panel or skeleton - all without the need to split the screen and invoke ISPF Test.

Note there are limitations to the Skeleton testing as variables and imbed tables may not be available.

Because of the use of ALTLIB or LIBDEF the member being tested will be able to find subroutines or other ISPF Panels providing they reside within the data set being edited thus allowing an entire package to be developed, updated, and/or tested, in less critical libraries.

The type of member being edited is dynamically determined with a default of REXX Exec assumed if all the tests fail. The tests include:

- The data set suffix
 - Assembler: ASM ASSEM
 - CLIST: CLIST, SYSPROC, CMDPROC
 - REXX: EXEC, REXX, SYSEXEC
 - Panel: PANEL, PANELS, ISPPLIB
 - Skels: SKEL, SKELS, ISPSLIB
- CLIST: Look for PROC followed by a number on record 1
- REXX: Look for the word REXX in record 1
- Panel: Look for any of these in record 1
-)ATTR)PANEL)CCSID)PROC)BODY)INIT)REINIT ..PREP:
- JCL: First record starts with //

This provides a very easy method for iterative testing and updating of the member until the member works as desired.

Usage

Use TRYIT from any ISPF Edit command line while editing a CLIST, REXX Exec, or ISPF Panel. The syntax is:

TRYIT optional-parms

The optional-parms are:

? to display the ISPF Tutorial

For CLISTs and REXX Execs any parameters that the member being edited would need to have passed to it.

For Assembler there are no supported options currently.

For ISPF Panels one, or more, of the following:

- APPL followed by a 1 to 4-character application id to be used when SEL is specified to select the panel
- POP will cause the panel to be displayed as a popup
 - optionally POP may be followed by a row and column to be used for the popup
- SEL will Select the panel instead of just Displaying it
- TUT will display the panel as an ISPF Tutorial
- default is to just display the panel

Example:

```
Command ==>tryit opt1 opt2
```

Upon completion of the processing of the member a message is displayed in the upper right with a return code or short message with a long message, available by pressing PF1 (Help), with more information. For CLISTs and REXX Execs the return code is whatever the CLIST or REXX Exec return while for ISPF Panels the message is either a zero-return code if there are no problems or the short and long error messages generated by the ISPF Display, Selection, or Tutorial services.

For Assembler programs upon completion of the assembly the assembly listing will be displayed using ISPF browse. If a load library is specified under the link edit options, then after the linkedit the linkedit listing will also be displayed using ISPF Browse. The linkedit will only occur if the assembly completes with a return code less than 8.

Download from <http://lbdsoftware.com/ispftools.html>

Other Tools of Note (meaning they are worth checking out)

These are useful tools that you should consider installing. All are open source (meaning free) and each will help with your application development in some way:

PDS (the Swiss Army Knife of Utilities)

Get this on the CBTTape Site in File 182 – www.cbttape.org – and be sure to check the updates page as a new update may have been released after the most recent CBTTape 'Tape' was cut. This utility, which works in native TSO, includes an extensive ISPF dialog and more sub-commands that you will ever be able to remember. One of the many powerful features is the ability to subset the list of members by a wide variety of different criteria, and then to operate on that member group using any of the dozens of sub-commands.

PDSEGEN

This is an excellent ISPF dialog that fully supports working with PDSE Version 2 Member Generations. While not as extensive in the sub-command area as PDS, this tool supports over almost 2 dozen line commands and over a dozen primary commands. Support is included to copy a PDSE, with all generations, from one PDSE to another. Another capability currently not supported by IBM is the ability to unload a PDSE with all generations and then reload it. Note that PDSEGEN does not support load libraries, but then why would anyone use member generations in a load library since there is no way to access a generation other than the base (generation 0) module.

ISPF CMD

This tool can be found at www.lbdsoftware.com, or in File 312 at www.cbttape.org, and is used to add your own ISPF commands to the sites ISPF command table dynamically. This can be very helpful if you need to add a command that your ISPF sysprog is not willing to install, or to install something that is unique for you.

Here are two useful commands to add using ISPF CMD:

```
"%ispfcmds * ivar      0 select cmd(%rxivar &zparm)" ,  
  "\ Display ISPF Variable"  
"%ispfcmds * rvar      0 select cmd(%rxrvar &zparm)" ,  
  "\ Display REXX Variable"
```

And the code behind these are:

RXIVAR and RXRVAR

IVAR:

```
/* ----- rexx----- */  
* IVAR - display the contents of the ispf variable that is *  
* passed as an argument. *  
* ----- */  
arg var  
address ispexec 'vget ('var zapplid)'  
say 'Requested variable' var 'in ISPF Applid' zapplid 'value is:'  
say ' '  
interpret 'str ='var  
say '>'str'<'
```

And **REXXVAR**:

```
/* ----- REXX ----- *
| Display any REXX variable, or evaluate and display any |
| REXX expression. |
* ----- */
parse arg opt
interpret say opt
```

Then from any ISPF command line enter **IVAR** or **RVAR** thus:

```
Menu Utilities Compilers Options Status Help
-----
                                ISPF Primary Option Menu
Option ==> ivar zuser
                                           More:
0 Settings      Terminal and user parameters
```

And get:

```
Requested variable ZUSER in ISPF Applid ISR value is:
>LBDYCK<
***
```

Or:

```
Menu Utilities Compilers Options Status H
-----
                                ISPF Primary Option M
Option ==> rvar sysvar('sysuid')
                                           More:
0 Settings      Terminal and user parameters
```

And get:

```
LBDYCK
***
```

Useful Websites

CBT Tape: www.cbttape.org

This is a repository of 100's of collections of tools contributed over decades by individuals from all over the world. The vast majority come with source and can be used 'out of the box' or as examples to learn from.

Lionel Dyck's site: www.lbdsoftware.com

This site has many of the tools referenced in this document and many others. Primarily for use with ISPF and focused on improving the productivity of both end users and systems programmers.

The Rexx Language Association: www.rexxla.org

The Rexx Language Association (RexxLA) is an independent, non-profit organization dedicated to promoting the use and understanding of the Rexx programming language.

Rexx Tutorial: <https://www.tutorialspoint.com/rexx/>

Learn Rexx for absolute beginners.